

Konventionelle Programmierung

Nr.	Titel	PGM-Nr.
	Bohrbearbeitung	
1	Zwei Lochzeilen	7280
2	Regelloses Bohrbild, 3 Werkzeuge	7153
3	Regelmäßige Lochfläche	7109
4	Regelmäßige Lochfläche versetzt	7426
5	Vier doppelte Lochkreise, 3 Werkzeuge	7139
6	Große Lochplatte	7411
	2D-Fräsen	
7	Einlegerahmen	72810
8	Doppelte L-Form 20er-Zyklen	7172
9	Mengenlehre innen mit Q-Parametern	761541
10	Mengenlehre innen und außen mit Q-Param.	761591
	Freie Kontur-Programmierung	
11	FK-Telefon	75103
12	FK Hantel	75116
13	FK Hammer elementar	75119
14	FK Hammer, Insel mit 20er Zyklen	7173
15	FK Micky Maus elementar	71750
16	FK Micky Maus mit 20er Zyklen	7175
17	FK T-Kontur mit 20er Zyklen	7174
18	FK Kamm elementar, unkorrigiert	75108
19	FK Kamm mit 20er Zyklen	7176
20	FK Kamm als Insel original und in Y gespiegelt	75122
	„Neue“ Funktionen	
21	Mehrfache Verwendung einer Insel	7177




Inhaltsverzeichnis Fortgeschrittenen-Kurs TNC 4xx

Nr.	Titel	PGM-Nr.
	3D-Fräsen	
22	Kniehebelgelenk, 1 Werkzeug	761260
23	Kniehebelgelenk, 2 Werkzeuge	761261
24	Kegelstumpf stehend, 2D, außen, horizontal	7162
25	Kegelstumpf stehend, 3D, außen	7282
26	Kegelstumpf innen, 3D, elementar	7283
27	Kegelstumpf innen, 3D, aus dem Vollen	7284
28	Kugel innen, 2D, horizontal	7322
29	Kugel innen, 2D, horizontal mit Taschen	7323
30	Kugel außen, 2D, horizontal	72841
31	Kugel außen, 3D, vertikal, 1 Schnitt	76130
32	Kugel außen, 3D, vertikal, 2 Schnitte	76131
33	Gewindefräsen dreigängig, 2 Zustellungen	7164

Q-Parameter-Programmierung

Nr.	Titel	PGM-Nr.
	Fräsbearbeitung	
34	Umriß fräsen in mehreren Z-Zustellungen	7285
35	Rechtecktasche fertig fräsen	7286
36	Kreis-Zapfen umfräsen	7288
	Bohrbearbeitung	
37	1 Lochkreis einfach	7416
38	6 Lochkreise 1 WZ	74160
39	6 Lochkreise 3 WZ, 3 Zyklen	74161
40	Regelmäßige Lochfläche	746910





Schema:

Vorbereitung

```

BLK FORM
TOOL DEF
TOOL CALL 1 Z S3000

CYCL DEF 1.0 TIEFBOHREN

L X... Y... R0 F9999 M3
L Z...

① CALL LBL 1

L X... Y...
L Z...

② CALL LBL 1
    
```

Rohteil
 Erster Werkzeug-
 Aufruf
 Tiefbohrzyklus
 Vorposition
 Vorposition

Freifahren, Ende


```
L Z... M2
```

UP1

```

LBL 1
L IX+20 M99
CALL LBL 1 REP 3/3
LBL 0
    
```



**HAUPTPROGRAMM**

```

0 BEGIN PGM 7280 MM
1 ..... 2 LOCHZEILEN
2 BLK FORM 0.1 Z X+0 Y+0 Z-30
3 BLK FORM 0.2 X+100 Y+60 Z+0
4 TOOL DEF 1 L+0 R+2,5
5 TOOL CALL 1 Z S3000
6 CYCL DEF 1.0 TIEFBOHREN
7 CYCL DEF 1.1 ABST -2
8 CYCL DEF 1.2 TIEFE -22
9 CYCL DEF 1.3 ZUSTLG -10
10 CYCL DEF 1.4 V.ZEIT 0
11 CYCL DEF 1.5 F150

12 L X+5 Y+45 R0 F9999 M3
13 L Z+2
14 CALL LBL 1

15 L X+10 Y+15
16 CALL LBL 1

Freifahren, Ende 17 L Z+20 R0 F9999 M2

```

UP1



```

18 LBL 1
19 L IX+20 F9999 M99
20 CALL LBL 1 REP 3/3
21 LBL 0
22 END PGM 7280 MM

```

**Aufgabe: Unregelmäßiges Bohrbild,
mit 3 Werkzeugen und 3 Zyklen**

Programm(e): _____



- Werkzeuge:**
- NC-Anbohrer
 - Spiralbohrer
 - Gewindebohrer

Schema: **Unregelmäßiges Bohrbild,
mit 3 Werkzeugen und 3 Zyklen**

Vorbereitung	<pre>BLK FORM TOOL DEF TOOL 1... LZ... M6</pre>	<p>Rohteil Werkzeugliste Erster Werkzeug- Aufruf</p> <p>Werkzeugwechsel</p>
Bearbeitungen Zentrieren	<pre>CYCL DEF, TIEFE = ZUSTLG. CALL LBL 1 STOP M6</pre>	<p>Tiefbohrzyklus</p> <p>Bohrbild aufrufen</p> <p>Werkzeugwechsel</p>
Bohren	<pre>TOOL 2... CYCL DEF, TIEFE ≠ ZUSTLG. CALL LBL 1 STOP M6</pre>	<p>Tiefbohrzyklus</p> <p>Bohrbild aufrufen</p> <p>Werkzeugwechsel</p>
Gewindebohren	<pre>TOOL 3... CYCL DEF CALL LBL 1</pre>	<p>Gewindebohrzyklus</p> <p>Bohrbild aufrufen</p>
Freifahren, Ende	<pre>LZ... M2</pre>	
Bohrbild, UP1	<pre>LBL 1 LX... Y... LZ... M89 LX... Y... LX... Y... M99 LZ... LX... Y... LBL 0</pre>	<p>1. Position links unten Sicherheitsabstand und modaler Zyklusaufruf</p> <p>Weitere Positionen Bohrung: automatisch</p> <p>Letzte Position Bohrung: Zyklusaufruf satzweise</p> <p>Wechselposition</p>



Schema:

Einsatz der Q-Parameter beim unregelmäßigen Bohrbild

BLK-FORM

Zentrieren

```
TOOL DEF 1 L0 R3,5  
TOOL CALL ...  
Q1 = +2  
Q2 = +3  
  
CYCL DEF ... TIEFBOHREN  
ABST -Q1  
TIEFE -Q2 ...  
  
CALL LBL 1
```

Abstand
Tiefe

Bohren

```
TOOL DEF 2 L0 R2,5  
TOOL CALL ...  
Q2 = 20  
  
CALL LBL 1
```

Zyklustyp bleibt
erhalten!

Gewindebohren

```
TOOL DEF 3 L0 R3,0  
TOOL CALL ...  
Q1 = 6  
Q2 = 15  
CYCL DEF ... GEWINDEBOHREN  
ABST -Q1  
TIEFE -Q2 ...  
  
CALL LBL 1
```

Freifahren, Ende

L Z100 M2

Rücksprung

Bohrbild, UP

```
LBL 1  
  
L X... Y... R0 F9999 M3  
L ZQ1 M89  
L X... Y...  
L X... Y... M99  
L Z50  
  
LBL 0
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C04



G7

Lösung:

Unregelmäßiges Bohrbild, mit 3 Werkzeugen und 3 Zyklen



HAUPTPROGRAMM

```
0 BEGIN PGM 7153 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-ANBOHRER
4 TOOL DEF 2 L+0 R+2,5 ..... SPIRALBOHRER
5 TOOL DEF 3 L+0 R+3 ..... GEWINDEBOHRER
```

Zentrieren

```
6 TOOL CALL 1 Z S800 ..... NC-ANBOHRER
7 L Z+100 R0 F9999 M6 ..... WERKZEUGWECHSEL
8 CYCL DEF 1.0 TIEFBOHREN
9 CYCL DEF 1.1 ABST-2
10 CYCL DEF 1.2 TIEFE-2
11 CYCL DEF 1.3 ZUSTLG-2
12 CYCL DEF 1.4 V.ZEIT0
13 CYCL DEF 1.5 F200
14 CALL LBL 1 ..... BOHRBILD AUFRUFEN
15 STOP M6 ..... WERKZEUGWECHSEL
```

Bohren

```
16 TOOL CALL 2 Z S500 ..... SPIRALBOHRER
17 CYCL DEF 1.0 TIEFBOHREN
18 CYCL DEF 1.1 ABST-2
19 CYCL DEF 1.2 TIEFE-20
20 CYCL DEF 1.3 ZUSTLG-10
21 CYCL DEF 1.4 V.ZEIT0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... BOHRBILD AUFRUFEN
24 STOP M6 ..... WERKZEUGWECHSEL
```

Gewindebohren

```
25 TOOL CALL 3 Z S500 ..... GEWINDEBOHRER
26 CYCL DEF 2.0 GEWINDEBOHREN
27 CYCL DEF 2.1 ABST-2
28 CYCL DEF 2.2 TIEFE-15
29 CYCL DEF 2.3 V.ZEIT0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... BOHRBILD AUFRUFEN
```

Freifahren, Ende

```
32 L Z+100 M2
```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



7153/3



Lösung:

Unregelmäßiges Bohrbild, mit 3 Werkzeugen und 3 Zyklen

UP1, BOHRBILD

```
33 LBL 1
34 L X+10 Y+20 R0 F9999 M13
35 L Z+2 M89 ..... ERSTER ZYKL.-RUF MODAL
36 L X+40 Y+30
37 L X+90 Y+10
38 L Y+40
39 L X+50 Y+80
40 L Y+50
41 L X+30 Y+40
42 L X+20 Y+80
43 L X+10 Y+90
44 L X+70 M99 ..... LETZTER ZYKL.-RUF SATZWEISE
45 L Z+100 M9 ..... WECHSELPOSITION
46 L X-20 Y-20
47 LBL 0
48 END PGM 7153 MM
```





Arbeitsweise: • Zeilen mäanderförmig abfahren

Werkzeuge:

- NC-Anbohrer
- Spiralbohrer
- Gewindebohrer

Schema: **Regelmäßige Lochfläche**

Vorbereitung

<i>BLK FORM</i> <i>TOOL 1.../TOOL 2...</i>

Rohteil
Ggf. Werkzeugliste

Bearbeitungen
Zentrieren

<i>TOOL 1...</i> <i>CYCL DEF, TIEFE = ZUSTLG.</i>
--

Tiefbohrzyklus

<i>CALL LBL 1</i>

Bohrbild aufrufen

<i>L Z... M6</i>

Werkzeugwechsel

Bohren

<i>TOOL 2...</i> <i>CYCL DEF, TIEFE ≠ ZUSTLG.</i>
--

Tiefbohrzyklus

<i>CALL LBL 1</i>

Bohrbild aufrufen

<i>L Z... M6</i>

Werkzeugwechsel

Gewindebohren

<i>TOOL 3...</i> <i>CYCL DEF</i>

Gewindebohrzyklus

<i>CALL LBL 1</i>


Bohrbild aufrufen

Freifahren, Ende

<i>L Z... M2</i>



Mäanderförmiges Abfahren



Bohrbild, UP1

LBL 1	
<i>L X... Y... R0 F9999 M3</i> <i>L Z... M8</i>	
LBL 2	
<i>L IX... M99</i>	
LBL 3	
<i>L IY... M99</i>	
<i>CALL LBL 3 REP...</i>	
<i>L IX... M99</i>	
LBL 4	
<i>L IY... M99</i>	
<i>CALL LBL 4 REP...</i>	
<i>CALL LBL 2 REP...</i>	
LBL 0	

Absolute Vorposition
Sicherheitsabstand

Sprungmarke
inkrem. Bereich

Quer-Schritt und
Bohrung

Sprungmarke


Längsbewegung
(aufwärts)

Quer-Schritt und
Bohrung

Sprungmarke

Längsbewegung
(abwärts)

Restliche Spalten



HAUPTPROGRAMM

```

0 BEGIN PGM 7109 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-ANBOHRER
4 TOOL DEF 2 L+0 R+2,5 ..... SPIRALBOHRER
5 TOOL DEF 3 L+0 R+3 ..... GEWINDEBOHRER
    
```

Zentrieren

```

6 TOOL CALL 1 Z S2500 ..... NC-ANBOHRER
7 L Z+100 R0 F9999 M6 ..... WERKZEUGWECHSEL
8 CYCL DEF 1.0 TIEFBOHREN
9 CYCL DEF 1.1 ABST -2
10 CYCL DEF 1.2 TIEFE -3
11 CYCL DEF 1.3 ZUSTLG -3
12 CYCL DEF 1.4 V.ZEIT 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1 ..... BOHRBILD AUFRUFEN
15 L Z+100 M6 ..... WERKZEUGWECHSEL
    
```

Bohren

```

16 TOOL CALL 2 Z S1500
17 CYCL DEF 1.0 TIEFBOHREN
18 CYCL DEF 1.1 ABST -2
19 CYCL DEF 1.2 TIEFE -20
20 CYCL DEF 1.3 ZUSTLG -20
21 CYCL DEF 1.4 V.ZEIT 0
22 CYCL DEF 1.5 F50
23 CALL LBL 1 ..... BOHRBILD AUFRUFEN
24 L Z+100 M6 ..... WERKZEUGWECHSEL
    
```

Gewindebohren

```

25 TOOL CALL 3 Z S500
26 CYCL DEF 2.0 GEWINDEBOHREN
27 CYCL DEF 2.1 ABST -2
28 CYCL DEF 2.2 TIEFE -15
29 CYCL DEF 2.3 V.ZEIT 0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... BOHRBILD AUFRUFEN
    
```

Freifahren, Ende

```

32 L Z+100 M2
    
```





UP1, Bohrbild

33	LBL 1	BOHRBILD
34	L X+5 Y+20 R0 F9999 M3	VORPOSITION
35	L Z+2 R0 M8	
36	LBL 2	
37	L IX+10 M99	QUER-SCHRITT
38	LBL 3	
39	L IY+8 M99	Y-SCHRITT AUFWAERTS
40	CALL LBL 3 REP 7/7	SPALTE AUFWAERTS
41	L IX+10 M99	QUER-SCHRITT
42	LBL 4	
43	L IY-8 M99	Y-SCHRITT ABWAERTS
44	CALL LBL 4 REP 7/7	SPALTE ABWAERTS
45	CALL LBL 2 REP 3/3	RESTLICHE SPALTEN
46	LBL 0	
47	END PGM 7109 MM	



Aufgabe: Regelmäßig versetzte Lochfläche, inclusive Gewindebohren

Programm(e): _____



Arbeitsweise: • Zeilen mäanderförmig abfahren


Werkzeuge:

- NC-Anbohrer
- Spiralbohrer
- Gewindebohrer

Schema:

Regelmäßig versetzte Lochfläche, inclusive Gewindebohren

Mäanderförmiges Abfahren



Linkes Ende der
kurzen Zeile =
Vorposition für
die nächste
lange Zeile

Erste Vorposition
liegt unter der
ersten langen
Zeile

Vorbereitung

```
BLK FORM
TOOL DEF
TOOL 1...
```

Rohteil
Werkzeugliste
Erster Werkzeug-
Aufruf

```
L Z... .. M6
```

Werkzeugwechsel

Bearbeitungen

Zentrieren

```
CYCL DEF, TIEFE = ZUSTLG.
```

Tiefbohrzyklus

```
CALL LBL 1
```

Bohrbild aufrufen

```
STOP M6
```

Werkzeugwechsel

Bohren

```
TOOL 2...
CYCL DEF, TIEFE ≠ ZUSTLG.
```

Tiefbohrzyklus

```
CALL LBL 1
```

Bohrbild aufrufen

```
STOP M6
```

Werkzeugwechsel

Gewindebohren

```
TOOL 3...
CYCL DEF
```

Gewindebohrzyklus

```
CALL LBL 1
```

Bohrbild aufrufen

Bearbeitungsende

```
STOP M2
```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C05



7426/2

Schema:

Regelmäßig versetzte Lochfläche, inclusive Gewindebohren


Bohrbild, UP1

<i>LBL 1</i>		
<i>L X... Y... L Z...</i>		Absolute Vorposition Sicherheitsabstand
<i>LBL 2</i>		Sprungmarke inkrem. Bereich
<i>L IX... IY... M99</i>		Quer-Schritt und Bohrung
<i>LBL 3</i>		Sprungmarke
<i>L IX... M99</i>		Längsbewegung (nach rechts)
<i>CALL LBL 3 REP...</i>		
<i>L IX... IY... M99</i>		Quer-Schritt und Bohrung
<i>LBL 4</i>		Sprungmarke
<i>L IX... M99</i>		Längsbewegung (nach links)
<i>CALL LBL 4 REP...</i>		
<i>CALL LBL 2 REP...</i>		Restliche Zeilen
<i>L Z... L X... Y... ..</i>		Wechselposition
<i>LBL 0</i>		



Lösung:

Regelmäßig versetzte Lochfläche, inclusive Gewindebohren



HAUPTPROGRAMM

```

0 BEGIN PGM 7426 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-ANBOHRER
4 TOOL DEF 2 L+0 R+2,5 ..... SPIRALBOHRER
5 TOOL DEF 3 L+0 R+3 ..... GEWINDEBOHRER

```

Zentrieren

```

6 TOOL CALL 1 Z S800 ..... NC-ANBOHRER
7 L Z+100 R0 F9999 M6 ..... WERKZEUGWECHSEL
8 CYCL DEF 1.0 TIEFBOHREN
9 CYCL DEF 1.1 ABST -2
10 CYCL DEF 1.2 TIEFE -2
11 CYCL DEF 1.3 ZUSTLG -2
12 CYCL DEF 1.4 V.ZEIT 0
13 CYCL DEF 1.5 F200
14 CALL LBL 1 ..... BOHRBILD AUFRUFEN
15 STOP M6 ..... WERKZEUGWECHSEL

```

Bohren

```

16 TOOL CALL 2 Z S500 ..... SPIRALBOHRER
17 CYCL DEF 1.0 TIEFBOHREN
18 CYCL DEF 1.1 ABST -2
19 CYCL DEF 1.2 TIEFE -20
20 CYCL DEF 1.3 ZUSTLG -10
21 CYCL DEF 1.4 V.ZEIT 0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... BOHRBILD AUFRUFEN
24 STOP M6 ..... WERKZEUGWECHSEL

```

Gewindebohren

```

25 TOOL CALL 3 Z S500 ..... GEWINDEBOHRER
26 CYCL DEF 2.0 GEWINDEBOHREN
27 CYCL DEF 2.1 ABST -2
28 CYCL DEF 2.2 TIEFE -15
29 CYCL DEF 2.3 V.ZEIT 0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... BOHRBILD AUFRUFEN

```

Freifahren, Ende

```

32 STOP M2

```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



7426/4

Lösung:

Regelmäßig versetzte Lochfläche, inclusive Gewindebohren

UP1, BOHRBILD

33 LBL 1
34 L X+25 Y+0 R0 F9999 M3 VORPOSITION
35 L Z+2 R0 M8

36 LBL 2 QUER-SCHRITT, BOHREN
37 L IX-5 IY+10 R0 M99

38 LBL 3 ZEILE NACH RECHTS
39 L IX+10 R0 M99
40 CALL LBL 3 REP 6/6

41 L IX-5 IY+10 R0 M99 QUER-SCHRITT

42 LBL 4 ZEILE NACH LINKS
43 L IX-10 R0 M99
44 CALL LBL 4 REP 5/5



45 CALL LBL 2 REP 3/3 RESTLICHE ZEILEN

46 L Z+100 R0 M9 WECHSELPOSITION
47 L X-20 Y-20 R0
48 LBL 0
49 END PGM 7426 MM



Aufgabe: **Vier doppelte Lochkreise,
mit 3 WZ, 3 Zyklen,
inclusive Gewindebohren**

Programm(e): _____



- Werkzeuge:**
- NC-Anbohrer
 - Spiralbohrer
 - Gewindebohrer

BLK-FORM

Zentrieren

*TOOL DEF / TOOL CALL 1 ...
CYCL DEF ...*

CALL LBL 1

Bohren

*TOOL DEF / TOOL CALL 2 ...
CYCL DEF ...*

CALL LBL 1

Gewindebohren

*TOOL DEF / TOOL CALL 3 ...
CYCL DEF ...*

CALL LBL 1

Freifahren, Ende

L Z100 M2

**UP1,
Kreismitten und
UP-Aufrufe**

LBL 1

CC X ... Y ...

CALL LBL 2

CC X ... Y ...

CALL LBL 2

CC X ... Y ...

CALL LBL 2

LBL 0

**UP2,
Verfahrbewegungen
und Bohrungen**

LBL 2

*LP PR ... PA ... M3
L Z2 M99*

LBL 3

•

•

LBL 4

•

•

LBL 0

Innerer Lochkreis

Äußerer Lochkreis

Schema:	Vier doppelte Lochkreise, mit 3 WZ, 3 Zyklen, inclusive Gewindebohren				
Vorbereitung	<table border="1"> <tr> <td><i>BLK FORM</i> <i>TOOL DEF</i> <i>TOOL 1...</i></td> </tr> <tr> <td><i>LZ... M6</i></td> </tr> </table>	<i>BLK FORM</i> <i>TOOL DEF</i> <i>TOOL 1...</i>	<i>LZ... M6</i>	Rohteil Werkzeugliste Erster Werkzeug- Aufruf Werkzeugwechsel	
<i>BLK FORM</i> <i>TOOL DEF</i> <i>TOOL 1...</i>					
<i>LZ... M6</i>					
Bearbeitungen Zentrieren	<table border="1"> <tr> <td><i>CYCL DEF, TIEFE = ZUSTLG.</i></td> </tr> <tr> <td><i>CALL LBL 1</i></td> </tr> <tr> <td><i>LZ... M6</i></td> </tr> </table>	<i>CYCL DEF, TIEFE = ZUSTLG.</i>	<i>CALL LBL 1</i>	<i>LZ... M6</i>	Tiefbohrzyklus Bohrbild aufrufen Werkzeugwechsel
<i>CYCL DEF, TIEFE = ZUSTLG.</i>					
<i>CALL LBL 1</i>					
<i>LZ... M6</i>					
Bohren	<table border="1"> <tr> <td><i>TOOL 2...</i> <i>CYCL DEF, TIEFE ≠ ZUSTLG.</i></td> </tr> <tr> <td><i>CALL LBL 1</i></td> </tr> <tr> <td><i>LZ... M6</i></td> </tr> </table>	<i>TOOL 2...</i> <i>CYCL DEF, TIEFE ≠ ZUSTLG.</i>	<i>CALL LBL 1</i>	<i>LZ... M6</i>	Tiefbohrzyklus Bohrbild aufrufen Werkzeugwechsel
<i>TOOL 2...</i> <i>CYCL DEF, TIEFE ≠ ZUSTLG.</i>					
<i>CALL LBL 1</i>					
<i>LZ... M6</i>					
Gewindebohren	<table border="1"> <tr> <td><i>TOOL 3...</i> <i>CYCL DEF</i></td> </tr> <tr> <td><i>CALL LBL 1</i></td> </tr> </table>	<i>TOOL 3...</i> <i>CYCL DEF</i>	<i>CALL LBL 1</i>	Gewindebohrzyklus Bohrbild aufrufen	
<i>TOOL 3...</i> <i>CYCL DEF</i>					
<i>CALL LBL 1</i>					
Freifahren, Ende	<i>LZ... M2</i>				



Schema: **Vier doppelte Lochkreise,
mit 3 WZ, 3 Zyklen,
inclusive Gewindebohren**





Kreiszentren, UP1

<i>LBL 1</i>	
<i>CC X... Y...</i>	
<i>CALL LBL 2</i>	
<i>CC X... Y...</i>	
<i>CALL LBL 2</i>	
<i>...</i>	
<i>LBL 0</i>	

Mittelpunkt links unten
Lochkreis aufrufen
Restliche Mittelpunkte
Restliche Lochkreise aufrufen

Ende UP1

**Bohrbild, UP2
Lochkreis**

<i>LBL 2</i>	
<i>LP PR... PA... R0 F9999 M13 LZ... M99</i>	
<i>LBL 3</i>	
<i>LP PR... IPA... M99</i>	
<i>CALL LBL 3 REP...</i>	
<i>LP PR... M99</i>	
<i>LBL 4</i>	
<i>LP PR... IPA... M99</i>	
<i>CALL LBL 4 REP...</i>	
<i>LBL 0</i>	


Bohrpositionen
Innerer Kreis
Sicherheitsabstand
und erste Bohrung
Sprungmarke
Restliche
Bohrpositionen
Äußerer Kreis,
erste Bohrung
Sprungmarke
Restliche
Bohrpositionen

Ende UP2



Lösung:

Vier doppelte Lochkreise, mit 3 WZ, 3 Zyklen, inclusive Gewindebohren



HAUPTPROGRAMM

```

0 BEGIN PGM 7139 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-ANBOHRER
4 TOOL DEF 2 L+0 R+2,5 ..... SPIRALBOHRER
5 TOOL DEF 3 L+0 R+3 ..... GEWINDEBOHRER
6 L Z+100 R0 F9999 M6 ..... WERKZEUGWECHSEL

```

Zentrieren

```

7 TOOL CALL 1 Z S2500 ..... NC-ANBOHRER
8 CYCL DEF 1.0 TIEFBOHREN
9 CYCL DEF 1.1 ABST -2
10 CYCL DEF 1.2 TIEFE -1,5
11 CYCL DEF 1.3 ZUSTLG -1,5
12 CYCL DEF 1.4 V.ZEIT 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1 ..... BOHRBILD AUFRUFEN
15 L Z+100 M6 ..... WERKZEUGWECHSEL

```

Bohren

```

16 TOOL CALL 2 Z S500
17 CYCL DEF 1.0 TIEFBOHREN
18 CYCL DEF 1.1 ABST -2
19 CYCL DEF 1.2 TIEFE -22
20 CYCL DEF 1.3 ZUSTLG -10
21 CYCL DEF 1.4 V.ZEIT 0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... BOHRBILD AUFRUFEN
24 L Z+100 M6 ..... WERKZEUGWECHSEL

```

Gewindebohren

```

25 TOOL CALL 3 Z S250
26 CYCL DEF 2.0 GEWINDEBOHREN
27 CYCL DEF 2.1 ABST -2
28 CYCL DEF 2.2 TIEFE -18
29 CYCL DEF 2.3 V.ZEIT 0
30 CYCL DEF 2.4 F250
31 CALL LBL 1 ..... BOHRBILD AUFRUFEN

```

Freifahren, Ende

```

32 L Z+100 M2

```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx D02



7139/4

Lösung:

**Vier doppelte Lochkreise,
mit 3 WZ, 3 Zyklen,
inclusive Gewindebohren**

UP1, Kreiszentren

33 LBL 1
34 CC X+25 Y+25 MITTELPUNKT LINKS UNTEN
35 CALL LBL 2 BOHRPOSITIONEN
36 CC X+75 Y+25 MITTELPUNKT RECHTS UNTEN
37 CALL LBL 2 BOHRPOSITIONEN
38 CC X+75 Y+75 MITTELPUNKT RECHTS OBEN
39 CALL LBL 2 BOHRPOSITIONEN
40 CC X+25 Y+75 MITTELPUNKT LINKS OBEN
41 CALL LBL 2 BOHRPOSITIONEN
42 LBL 0

UP1, Ende

**UP2, Lochkreis
selbst**

43 LBL 2 BOHRPOSITIONEN
44 LP PR+10 PA+0 M13 INNERER KREIS
45 L Z+2 M99 1. BOHRUNG

46 LBL 3 RESTLICHE BOHRUNGEN
47 LP IPA+90 M99
48 CALL LBL 3 REP 2/2


49 LP PR+18 M99 AEUSSERER KREIS

50 LBL 4 RESTLICHE BOHRUNGEN
51 LP IPA+45 M99
52 CALL LBL 4 REP 6/6

UP2, Ende

53 LBL 0
54 END PGM 7139 MM





- Arbeitsweise:**
- Bohrungen zentrieren, bohren, gewindebohren
 - Große Bohrung fräsen

- Werkzeuge:**
- NC-Anbohrer
 - Spiralbohrer
 - Gewindebohrer
 - Fräser R30

Schema: **Große Lochplatte**

Vorbereitung

<i>BLK FORM</i>
<i>TOOL 1.../TOOL 2...</i>

Rohteil
Ggf. Werkzeugliste

Bearbeitungen
Zentrieren

<i>TOOL 1... (R 2,0)</i>
<i>CYCL DEF, TIEFE = ZUSTLG.</i>
<i>CALL LBL 1</i>
<i>... M6</i>

Tiefbohrzyklus
Bohrbild aufrufen
Werkzeugwechsel

Bohren

<i>TOOL 2... (R 1,0)</i>
<i>CYCL DEF</i>
<i>CALL LBL 1</i>
<i>... M6</i>

Tiefbohrzyklus
Bohrbild aufrufen
Werkzeugwechsel

Gewindebohren

<i>TOOL 3... (R 1,5)</i>
<i>CYCL DEF</i>
<i>CALL LBL 1</i>
<i>... M6</i>

Gewindebohrzyklus
Bohrbild aufrufen
Werkzeugwechsel

Große Bohrung
ausfräsen

<i>TOOL 4... (R 30)</i>
<i>CYCL DEF</i>
<i>L X... Y...</i>
<i>L Z... .. M99</i>





Kreistaschen-Zyklus
Startposition und
Zyklus-Aufruf


Freifahren, Ende


L Z... .. M2



Bohrbild, UP1

<i>LBL 1</i>		
<i>CC...</i> <i>LP...</i>		Mittelpunkt Absolute Vorposition Ebene
<i>LZ...</i>		Sicherheitsabstand Z
<i>LBL 2</i>		
<i>LP IPA... .. M99</i>		Inkrem. Winkelschritt, Bohrung mit M99
<i>LBL 3</i>		
<i>LP IPR... .. M99</i>		Radiusschritt n. innen
<i>CALL LBL 3 REP...</i>		Weitere Radiusschritte
<i>LP IPA... .. M99</i>		Winkelschritt
<i>LBL 4</i>		
<i>LP IPR... .. M99</i>		Radiusschritt n. außen
<i>CALL LBL 4 REP...</i>		Weitere Radiusschritte
<i>CALL LBL 2 REP...</i>		Weitere Doppelzeilen
<i>LP IPA... ..</i>		Zur Zwischenposition
<i>CALL LBL 2 REP...</i>		Restl. Bohrgruppen
<i>LZ... .. M9</i> <i>LX... Y... ..</i>		Wechselposition
<i>LBL 0</i>		





HAUPTPROGRAMM

```

0 BEGIN PGM 7411 MM
1 BLK FORM 0.1 Z X-125 Y-130 Z-22
2 BLK FORM 0.2 X+125 Y+130 Z+0
3 TOOL DEF 1 L+0 R+2,0 ..... ZENTRIERBOHRER
4 TOOL DEF 2 L+0 R+1 ..... SPIRALBOHRER
5 TOOL DEF 3 L+0 R+1,5 ..... GEWINDEBOHRER
6 TOOL DEF 4 L+0 R+30 ..... FRAESER
    
```

Zentrieren

```

7 TOOL CALL 1 Z S3000
8 CYCL DEF 1.0 TIEFBOHREN
9 CYCL DEF 1.1 ABST -2
10 CYCL DEF 1.2 TIEFE -2
11 CYCL DEF 1.3 ZUSTLG -2
12 CYCL DEF 1.4 V.ZEIT 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1
15 L Z+20 R0 F9999 M6
    
```

Bohren

```

16 TOOL CALL 2 Z S3000
17 CYCL DEF 1.0 TIEFBOHREN
18 CYCL DEF 1.1 ABST -2
19 CYCL DEF 1.2 TIEFE -20
20 CYCL DEF 1.3 ZUSTLG -20
21 CYCL DEF 1.4 V.ZEIT 0
22 CYCL DEF 1.5 F200
23 CALL LBL 1
24 L Z+20 R0 F9999 M6
    
```



Gewindebohren

25 TOOL CALL 3 Z S500
 26 CYCL DEF 2.0 GEWINDEBOHREN
 27 CYCL DEF 2.1 ABST -2
 28 CYCL DEF 2.2 TIEFE -17
 29 CYCL DEF 2.3 V.ZEIT 0
 30 CYCL DEF 2.4 F250
 31 CALL LBL 1
 32 L Z+20 R0 F9999 M6

Kreistasche

33 TOOL CALL 4 Z S400
 34 CYCL DEF 5.0 KREISTASCHE
 35 CYCL DEF 5.1 ABST -2
 36 CYCL DEF 5.2 TIEFE -15
 37 CYCL DEF 5.3 ZUSTLG -5 F100
 38 CYCL DEF 5.4 RADIUS 75
 39 CYCL DEF 5.5 F500 DR-
 40 L X+0 Y+0 R0 M3
 41 L Z+2 M99

Freifahren, Ende

42 L Z+20 R0 F9999 M2

UP1, BOHRBILD

43 LBL 1
 44 CC X+0 Y+0 MITTELPUNKT
 45 LP PR+120 PA-6,666 R0 F9999 M3 1. BOHRUNG IN VORPOS.
 46 L Z+2 R0 M8 ANFAHREN

 47 LBL 2
 48 LP IPA+3,333 M99 WINKELSCHRITT

 49 LBL 3
 50 LP IPR-8 M99
 51 CALL LBL 3 REP 2/2 ZEILE RADIAL NACH INNEN


 52 LP IPA+3,333 M99 WINKELSCHRITT


 53 LBL 4
 54 LP IPR+8 M99
 55 CALL LBL 4 REP 2/2 ZEILE RADIAL NACH AUSSEN

 56 CALL LBL 2 REP 1/1 WEITERE DOPPELZEILEN

 57 LP IPA+4,666 LUECKE
 58 CALL LBL 2 REP 19/19 WEITERE 16ER-GRUPPEN
 WIEDERHOLEN

 59 LBL 0
 60 END PGM 7411 MM





- Arbeitsweise:**
- Innen ausräumen
 - An den Ecken der Einlegeflächen um den Fräserradius weiterfahren
 - An seitlicher Ausbuchtung kleine Konturstufe

- Werkzeuge:**
- Schrupp-Schlichtfräser R5

Schema: **Einlegerahmen für Platine mit M97 und M98**

Vorbereitung

```
BLK FORM
TOOL DEF
TOOL 1...

L Z... .. M6
```

Rohteil
Werkzeugliste
Werkzeug-Aufruf
Werkzeugwechsel

Bearbeitungen
Innen ausräumen

```
CYCL DEF
L X... Y...
L Z... .. M99
```

Taschenfräs-Zyklus
Startposition und
Zyklus-Aufruf

Einlegefläche

```
L X... Y... R0
L Z...
APPR... X... Y...

L X... M98

L X... Y...
L Y... M97

L X...
L Y...
L X... M97

L Y... M98

L X...
L X... M98

L X...
DEP... X... Y...
```

Vorposition
Tiefe
Kontur weich anfahren,
unten Mitte
In Ecke weiter um
Werkzeugradius

Kleine Konturstufe

Ausbuchtung
Ausbuchtung
Konturstufe zurück

Nächste Ecke

Nächste Ecke

Konturende
Kontur weich verlassen


Freifahren, Ende

```
L Z... .. M2
```

Wechselposition

Lösung:

Einlegerahmen für Platine mit M97 und M98



Ausräumen

```
0 BEGIN PGM 72810 MM
1 ..... EINLEGERAHM. MIT APPR/DEP
2 BLK FORM 0.1 Z X+0 Y+0 Z-25
3 BLK FORM 0.2 X+110 Y+110 Z+0
4 TOOL DEF 1 L+0 R+5
5 TOOL CALL 1 Z S2500
6 L Z+100 R0 F9999 M6
7 CYCL DEF 4.0 TASCHENFRAESEN
8 CYCL DEF 4.1 ABST -2
9 CYCL DEF 4.2 TIEFE -10
10 CYCL DEF 4.3 ZUSTLG -10 F150
11 CYCL DEF 4.4 X+82
12 CYCL DEF 4.5 Y+82
13 CYCL DEF 4.6 F1000 DR-
14 L X+55 Y+55 R0 M3
15 L Z+2 R0 M99
```

Umfräsen

```
16 L X+30 Y+30
17 L Z-5 R0 F100
18 APPR LCT X+30 Y+10 R5 RR F250
19 L X+10 M98
20 L X+10 Y+15
21 L Y+50 M97
22 L X+8
23 L Y+90
24 L X+10 M97
25 L Y+100 M98
26 L X+15
27 L X+100 M98
28 L Y+95
29 L Y+10 M98
30 L X+95
31 L X+30
32 DEP LCT X+30 Y+30 R5

33 L Z+100 R0 F9999 M2
34 END PGM 72810 MM
```






HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



72810/3





Schema: **Doppelte L-Form mit SL II-Zyklen**

Vorbereitung

Teilkonturen listen

```
BLK FORM
CYCL DEF 14.0 KONTUR
CYCL DEF 14.1 KONTURLABEL 1
CYCL DEF 20 KONTUR-DATEN
```

Generelle Daten für Fertigung (DEF-Zyklen)

Vorbohren

```
TOOL 1 . . .
CYCL DEF 21 VORBOHREN
:
CALL LBL 10
```

CALL-Zyklen

Räumen

```
TOOL 2 . . .
CYCL DEF 22 RAEUMEN
:
CALL LBL 10
```

Schlichten Tiefe

```
TOOL 3 . . .
CYCL DEF 23 SCHLICHTEN TIEFE
:
CALL LBL 10
```

Schlichten Seite

```
CYCL DEF 24 SCHLICHTEN SEITE
:
CALL LBL 10
```

Freifahren, Ende

```
L Z . . . M2
```

UP10

```
LBL 10

CYCL DEF 7.0 NULLPUNKT
CYCL CALL

CYCL DEF 7.0 NULLPUNKT

CYCL DEF 10.0 DREHUNG
CYCL DEF 10.1 ROT . . .
CYCL CALL

CYCL DEF 10.0 DREHUNG
CYCL DEF 10.1 ROT+0


LBL 0
```

Übergeordnetes UP

**Teilkonturen,
UP1 . . . UPn**

```
LBL 1 . . . LBL 0
:
LBL n . . . LBL 0
```





HAUPTPROGRAMM

```

0 BEGIN PGM 7172 MM
1 ..... DOPPEL-L-FORM MIT
2 ..... 20ER ZYKLEN
3 BLK FORM 0.1 Z X+0 Y+0 Z-22
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 CYCL DEF 14.0 KONTUR
6 CYCL DEF 14.1 KONTURLABEL 1
7 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-20 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN-OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+20 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
8 TOOL DEF 1 L+0 R+3
9 TOOL CALL 1 Z S2000
10 L Z+100 R0 F9999
11 L X-20 Y-20 R0 M3
12 CYCL DEF 21.0 VORBOHREN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q13=2 ..... AUSRAEUM-WERKZEUG
13 L Z+20 R0 F9999 M3
14 CALL LBL 10

15 TOOL DEF 2 L+0 R+4
16 TOOL CALL 2 Z S3150
17 STOP
18 CYCL DEF 22.0 RAEUMEN
  Q10=-10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
19 L Z+2 R0 F MAX M3
20 CALL LBL 10
    
```

21 L Z+100 R0 F MAX M6

22 TOOL DEF 3 L+0 R+3

23 TOOL CALL 3 Z S3150

24 CYCL DEF 23.0 SCHLICHTEN TIEFE

Q11=100 VORSCHUB TIEFENZUSTELLUNG

Q12=500 VORSCHUB RAEUMEN

25 CALL LBL 10

26 STOP

27 CYCL DEF 24.0 SCHLICHTEN SEITE

Q9=+1 DREHSINN

Q10=-10 ZUSTELL-TIEFE

Q11=100 VORSCHUB TIEFENZUSTELLUNG

Q12=500 VORSCHUB RAEUMEN

Q14=+0 AUFMASS SEITE

28 L Z+20 R0 F MAX M3

29 CALL LBL 10

Freifahren, Ende

30 L Z+100 R0 F MAX M2

UP10

31 LBL 10

32 CYCL DEF 7.0 NULLPUNKT

33 CYCL DEF 7.1 X+20

34 CYCL DEF 7.2 Y+12

35 CYCL CALL M3

36 CYCL DEF 7.0 NULLPUNKT

37 CYCL DEF 7.1 Y+50

38 CYCL DEF 7.2 X+30

39 CYCL DEF 10.0 DREHUNG

40 CYCL DEF 10.1 ROT+35

41 CYCL CALL

42 CYCL DEF 10.0 DREHUNG

43 CYCL DEF 10.1 ROT+0

44 LBL 0

UP10, Ende

UP1

45 LBL 1

46 L X+10 Y+0 RR

47 L X+0

48 RND R6

49 L Y+30

50 RND R6

51 L X+15

52 RND R6

53 L Y+15

54 L X+60

55 RND R6

56 L Y+0

57 RND R6

58 L X+10

59 LBL 0



UP1, Ende

60 END PGM 7172 MM



Aufgabe: **Mengenlehre innen mit Q-Parametern**

Programm(e): _____



Schema:

Verwendung von SL II-Zyklen für in der Tiefe gestaffelte Kontur

Vorbereitung

Schruppen

```
BLK FORM
TOOL 1... R3,5
CYCL DEF 20.0 KONTUR-DATEN
... = 0,4
... = 0
... = -2...
CYCL DEF 22.0 RAEUMEN...
```

Aufmaß Seite
Oberfläche
Sicherheits-Abstand

CALL LBL 10

Schlichten

```
TOOL 2... R2
CYCL DEF 24.0 SCHLICHTEN
SEITE
... = 0
```

Aufmaß Seite

CALL LBL 10

Ecken „Ausputzen“

```
TOOL 3... R1
```

CALL LBL 10

Freifahren, Ende

L Z... M2

UP10

LBL 10

1. Etage

```
L Z20... M3
CYCL DEF 14 KONTURLABEL 1/2
Q5 = 0
L X... Y... M99
```

Oberfläche

2. Etage

```
CYCL DEF 14 KONTUR-
LABEL 1/2/3
Q5 = -8
CYCL CALL
```

Oberfläche

3. Etage

```
CYCL DEF 14 KONTURLABEL 2/3
Q5 = -16
Q6 = -10
CYCL CALL
```

Oberfläche
S.-Abstand vergrößert

LBL 0

Teilkonturen,
UP1, UP2, UP3

LBL 1... LBL 0

Rechteck-Tasche

LBL 2... LBL 0

Kreis-Tasche

LBL 3... LBL 0

Ovale Insel




HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



761541/2



HAUPTPROGRAMM

```

0 BEGIN PGM 761541 MM
1 ..... VERGASERKAMMER INNEN 4XX
2 BLK FORM 0.1 Z X+0 Y+15 Z-25
3 BLK FORM 0.2 X+100 Y+95 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S3000
6 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-8 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+0,4 ..... AUFMASS SEITE
  Q4=+0 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN-OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+10 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
7 CYCL DEF 22.0 RAEUMEN
  Q10=-4 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=500 ..... VORSCHUB RAEUMEN
8 CALL LBL 10
9 STOP M6

10 TOOL DEF 2 L+0 R+2
11 TOOL CALL 2 Z S3000
12 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=+1 ..... DREHSINN
  Q10=-4 ..... ZUSTELL-TIEFE
  Q11=500 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=1000 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
13 CALL LBL 10
14 STOP M6

15 TOOL DEF 3 L+0 R+1
16 TOOL CALL 3 Z S4000
17 CALL LBL 10

18 L Z+50 R0 F9998 M2
    
```

Freifahren, Ende



UP10

19 LBL 10
 20 L Z+20 R0 F9999 M3
 21 CYCL DEF 14.0 KONTUR
 22 CYCL DEF 14.1 KONTURLABEL 1 /2
 23 FN 0: Q5 = +0 OBERFLAECHE
 24 L X+10 Y+25 R0 F9998 M99
 25 CYCL DEF 14.0 KONTUR
 26 CYCL DEF 14.1 KONTURLABEL 1 /2 /3
 27 FN 0: Q5 = -8 OBERFLAECHE
 28 CYCL CALL

 29 CYCL DEF 14.0 KONTUR
 30 CYCL DEF 14.1 KONTURLABEL 2 /3
 31 FN 0: Q5 = -16 OBERFLAECHE
 32 FN 0: Q6 = -10 ABSTAND
 33 CYCL CALL
 34 LBL 0

**UP1,
Rechteck-Tasche**

35 LBL 1
 36 L X+5 Y+20 RR
 37 L Y+75
 38 L X+60
 39 L Y+20
 40 L X+5
 41 LBL 0

UP2, Kreis-Tasche

42 LBL 2
 43 CC X+60 Y+55
 44 LP PR+35 PA+0 RR
 45 CP DR-
 46 LBL 0


UP3, Insel


47 LBL 3
 48 L X+55 Y+30 RL
 49 L X+25
 50 CT Y+50
 51 L X+55
 52 CT Y+30
 53 LBL 0
 54 END PGM 761541 MM



Aufgabe: **Mengenlehre innen und außen mit Q-Parametern**

Programm(e): _____





HAUPTPROGRAMM

```

0 BEGIN PGM 761591 MM
1 ..... VERGASERKAMMER INNEN U.
2 ..... AUSSEN, TNC 4XX
3 BLK FORM 0.1 Z X+0 Y+15 Z-25
4 BLK FORM 0.2 X+100 Y+95 Z+0
5 TOOL DEF 1 L+0 R+3,5
6 TOOL CALL 1 Z S3000
7 L Z+20 R0 F9998 M3
8 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-24 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+0 ..... AUFMASS SEITE
  Q4=+0 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN-OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+10 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
9 CYCL DEF 22.0 RAEUMEN
  Q10=-4 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=500 ..... VORSCHUB RAEUMEN
10 CYCL DEF 14.0 KONTUR
11 CYCL DEF 14.1 KONTURLABEL 4 /5 /6
12 CYCL CALL M3
13 CALL LBL 10
14 STOP M6

```

Räumen

```

15 TOOL DEF 2 L+0 R+2
16 TOOL CALL 2 Z S3000
17 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=+1 ..... DREHSINN
  Q10=-4 ..... ZUSTELL-TIEFE
  Q11=500 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=1000 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
18 CALL LBL 10
19 STOP M6

```

Schlichten



20 TOOL DEF 3 L+0 R+1
 21 TOOL CALL 3 Z S4000
 22 CALL LBL 10

23 L Z+50 R0 F9998 M2

UP

24 LBL 10
 25 CYCL DEF 14.0 KONTUR
 26 CYCL DEF 14.1 KONTURLABEL 1 /2
 27 FN 0: Q1 = -8 TIEFE
 28 FN 0: Q3 = +0,4 AUFMASS SEITE
 29 FN 0: Q5 = +0 OBERFLAECHE
 30 FN 0: Q6 = -2 SICHERHEIT
 31 CYCL CALL M3
 32 CYCL DEF 14.0 KONTUR
 33 CYCL DEF 14.1 KONTURLABEL 1 /2 /3
 34 FN 0: Q5 = -8 OBERFLAECHE
 35 CYCL CALL
 36 CYCL DEF 14.0 KONTUR
 37 CYCL DEF 14.1 KONTURLABEL 2 /3
 38 FN 0: Q5 = -16 OBERFLAECHE
 39 FN 0: Q1 = -10 ABSTAND
 40 CYCL CALL


Innenelemente


41 LBL 0
 42 LBL 1
 43 L X+5 Y+20 RR
 44 L Y+75
 45 L X+60
 46 L Y+20
 47 L X+5
 48 LBL 0
 49 LBL 2
 50 CC X+60 Y+55
 51 LP PR+35 PA+0 RR
 52 CP DR-
 53 LBL 0
 54 LBL 3
 55 L X+55 Y+30 RL
 56 L X+25
 57 CT Y+50
 58 L X+55
 59 CT Y+30
 60 LBL 0


Außenelemente

61 LBL 4
 62 L X-12 Y+0 RR
 63 L Y+105
 64 L X+110
 65 L Y+0
 66 L X-12
 67 LBL 0
 68 LBL 5
 69 L X+0 Y+15 RL
 70 L Y+80
 71 L X+55
 72 L Y+15
 73 L X+0
 74 LBL 0
 75 LBL 6
 76 CC X+60 Y+55
 77 LP PR+40 PA+0 RL
 78 CP DR-
 79 LBL 0
 80 END PGM 761591 MM









FK-Programm

```

0 BEGIN PGM 75103 MM
1 BLK FORM 0.1 Z X+0 Y+50 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S2800
5 L X+50 Y+50 R0 F9999 M13
6 L Z-5 F100
7 L X+50 Y+75 RL .....①
8 RND R5 F100

9 FC DR+ R25 CCX+50 CCY+50 F500 .....②
10 FCT DR- R14 .....③
11 FCT DR- R88 CCX+50 CCY+0 .....④
12 FCT DR- R14 .....⑤
13 FCT X+50 Y+75 DR+ R25 CCX+50 CCY+50 .....⑥
14 FSELECT 2

15 RND R5 F100
16 L X+50 Y+50 R0 M9

17 L Z+10
18 END PGM 75103 MM
    
```

Konvertiertes Programm



```


0 BEGIN PGM BOGEN MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3
4 TOOL CALL 1 Z S2800
5 L X+50 Y+50 R0 F9999 M13
6 L Z-5 F100
7 L X+50 Y+75 RL .....①
8 RND R5 F100

9 CC X+50 Y+50
10 C X+26,805 Y+59,3269 DR+
11 CC X+13,8158 Y+64,55
12 C X+6,9701 Y+76,7622 DR-
13 CC X+50 Y+0
14 C X+93,0299 Y+76,7622 DR-
15 CC X+86,1842 Y+64,55
16 C X+73,195 Y+59,3269 DR-
17 CC X+50 Y+50
18 C X+50 Y+75 DR+

19 RND R5 F100
20 L X+50 Y+50 R0 M9

21 L Z+10
22 END PGM BOGEN MM
    
```








```

0 BEGIN PGM 75116 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+4
4 TOOL CALL 1 Z S4000
5 L Z+2 R0 F MAX M3
6 L X+40 Y+30 R0 F2000
7 L Z-10 F100
8 L X+10 Y+30 RR F200
9 FPOL X+40 Y+30
10 FC DR- R30 CCX+40 CCY+30 F500
11 FL AN+60 PDX+40 PDY+30 D10
12 FSELECT 3
13 FC DR- R20 CCPR+55 CCPA+60
14 FSELECT 2
15 FL AN-120 PDX+40 PDY+30 D10
16 FSELECT 3
17 FC X+10 Y+30 DR- R30 CCX+40 CCY+30
18 FSELECT 2
19 L X+40 Y+30 R0
20 L Z+50 R0 F MAX M2
21 END PGM 75116 MM

```





```

0 BEGIN PGM 75119 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+85 Z+0
3 TOOL DEF 1 L+0 R+5
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+80 Y+70 F MAX M3
7 L Z-10 F100
8 CC X+30 Y+40
9 LP PR+37,5 PA+30 RL F500
10 FL LEN10 AN-60
11 FL AN-150
12 FCT DR+ R12,5
13 FLT AN-100
14 FCT DR- R11 CCX+30 CCY+15
15 FLT PDX+15 PDY+65 D11
16 FSELECT 1
17 FCT DR- R11 CCX+15 CCY+65
18 FLT AN-55
19 FCT DR+ R15
20 FPOL X+30 Y+40
21 FLT AN+30 PDX+30 PDY+40 D10
22 FSELECT 2
23 FL PR+37,5 PA+30 IAN+90
24 L X+80 Y+70 R0 F500
25 L Z+50 R0 F MAX M2
26 END PGM 75119 MM


```

Schema: **FK Hammer, Insel mit 20er-Zyklen,
mit einem Werkzeug**

Vorbereitung	<i>BLK FORM TOOL 1 . . .</i>	
Auflistung	<i>CYCL DEF 14 LABEL 1/2 CYCL DEF 20 KONTUR-DATEN CYCL DEF 22 RAEUMEN</i>	Teilkonturen Schnittdaten Schruppen
Vorpositionieren	<i>L X . . . Y . . . F9999 M3 CYCL CALL CYCL DEF 23 SCHLICHTEN TIEFE CYCL CALL CYCL DEF 24 SCHLICHTEN SEITE CYCL CALL</i>	Ebene (wegen Spannmitteln) Aufruf Schichten Tiefe Aufruf Schichten
Freifahren, Ende	<i>L Z100 M2</i>	

UP, Teilkonturen	<i>LBL 1</i>	Insel Reiner Umriß! Ohne An- und Wegfahren
	<i>L X . . . Y . . . RL : : :</i>	
	<i>LBL 0</i>	
	<i>LBL 2</i>	Tasche Reiner Umriß!
	<i>L X . . . Y . . . RR : : :</i>	
	<i>LBL 0</i>	





HAUPTPROGRAMM

```

0 BEGIN PGM 7173 MM
1 ..... HAMMERKOPF MIT 20ER ZYKL.
2 ..... BO6
3 BLK FORM 0.1 Z X+0 Y+0 Z-20
4 BLK FORM 0.2 X+70 Y+80 Z+0
5 TOOL DEF 5 L+0 R+5
6 TOOL CALL 5 Z S2500
7 CYCL DEF 14.0 KONTUR
8 CYCL DEF 14.1 KONTURLABEL 1/2
9 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-10 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+100 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
10 CYCL DEF 22.0 RAEUMEN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=500 ..... VORSCHUB RAEUMEN
11 L X+80 Y+70 R0 F9999 M3
12 CYCL CALL

13 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=500 ..... VORSCHUB RAEUMEN
14 CYCL CALL

15 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=500 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
16 CYCL CALL


Freifahren, Ende
17 L Z+100 R0 F MAX M2
    
```


UP

18 LBL 1
19 L X+30 Y+4 RL
20 FC DR- R11 CCX+30 CCY+15
21 FLT
22 FCT DR- R11 CCA+35 CCX+15 CCY+65
23 FLT
24 FCT DR+ R15
25 FLT AN+30 PDX+30 PDY+40 D10
26 FSELECT 2
27 FL LEN20 AN-60 PDX+30 PDY+40 D37,5
28 FL AN+210 PDX+30 PDY+40 D10
29 FCT DR+ R12,5
30 FLT AN-100
31 FCT X+30 DR- R11 CCX+30 CCY+15
32 FSELECT 2
33 LBL 0

34 LBL 2
35 L X+30 Y-15 RR
36 L X-8
37 L Y+90
38 L X+90
39 L Y-15
40 L X+30
41 LBL 0
42 END PGM 7173 MM








```

0 BEGIN PGM 71750 MM
1 ..... MICKY
2 BLK FORM 0.1 Z X-70 Y-90 Z-12
3 BLK FORM 0.2 X+70 Y+30 Z+0
4 TOOL DEF 1 L+0 R+5
5 TOOL CALL 1 Z S2500
6 L Z+10 R0 F9999 M3
7 L X-62 Y+0 RL
8 L Z-10 F150
9 FC DR- R22 CLSD+ CCX-40 CCY+0
10 FCT DR+ R38 CCX+0
11 FSELECT 1
12 FCT DR- R22 CCX+40 CCY+0
13 FCT DR+ R10
14 FLT AN-60 PDX+0 PDY+0 D0
15 FSELECT 3
16 FCT DR- R8
17 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
18 FLT X-40 AN+180
19 FL AN+90
20 FL AN+30
21 FCT DR+ R9 CCX-17
22 FLT X-30 PAR21 DP18
23 FSELECT 2
24 FL AN+90
25 FL PAR23 DP8
26 FSELECT 2
27 FCT DR+ R9 CCX-17
28 FCT X-62 Y+0 DR- R22 CLSD- CCX-40 CCY+0
29 FSELECT 2
30 L Z+10 F9999 M2
31 END PGM 71750 MM
    
```



HAUPTPROGRAMM

```

0 BEGIN PGM 7175 MM
1 ..... MICKY MIT 20ER ZYKLEN
2 BLK FORM 0.1 Z X-70 Y-90 Z-12
3 BLK FORM 0.2 X+70 Y+30 Z+0
4 TOOL DEF 5 L+0 R+5
5 TOOL CALL 5 Z S2500
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 KONTUR
8 CYCL DEF 14.1 KONTURLABEL 1/2
9 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-10 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+20 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
10 CYCL DEF 22.0 RAEUMEN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
11 L X-65 Y+0 M99

12 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
13 CYCL CALL

14 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
15 CYCL CALL

16 L Z+100 R0 F MAX M2
    
```

Freifahren, Ende

Lösung:

FK Micky Maus mit 20er-Zyklen

UP

```
17 LBL 1
18 L X-62 Y+0 RL
19 FC DR- R22 CLSD+ CCX-40 CCY+0
20 FCT DR+ R38 CCX+0
21 FSELECT 1
22 FCT DR- R22 CCX+40 CCY+0
23 FCT DR+ R10
24 FLT AN-60 PDX+0 PDY+0 D0
25 FSELECT 3
26 FCT DR- R8
27 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
28 FLT X-40 AN+180
29 FL AN+90
30 FL AN+30
31 FCT DR+ R9 CCX-17
32 FLT X-30 PAR31 DP18
33 FSELECT 2
34 FL AN+90
35 FL PAR33 DP8
36 FSELECT 2
37 FCT DR+ R9 CCX-17
38 FCT DR- R22 CLSD- CCX-40 CCY+0
39 FSELECT 2
40 LBL 0

41 LBL 2
42 L X-74 Y+20 RR
43 L Y+35
44 L X+74
45 L Y-100
46 L X-74
47 L Y+20
48 LBL 0
49 END PGM 7175 MM
```





HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C05



7175/4





```

0 BEGIN PGM 7174 MM
1 ..... T-KONTUR MIT 20ER ZYKLEN
2 ..... B06
3 BLK FORM 0.1 Z X-70 Y-100 Z-20
4 BLK FORM 0.2 X+70 Y+5 Z+0
5 TOOL DEF 1 L+0 R+6
6 TOOL CALL 1 Z S2000
7 L Z+20 R0 F9999 M3
8 CYCL DEF 14.0 KONTUR
9 CYCL DEF 14.1 KONTURLABEL 1 /2
10 CYCL DEF 20.0 KONTUR-DATEN
    Q1=-10 ..... FRAESTIEFE
    Q2=1 ..... BAHN-UEBERLAPPUNG
    Q3=+1 ..... AUFMASS SEITE
    Q4=+1 ..... AUFMASS TIEFE
    Q5=+0 ..... KOORDINATEN OBERFLAECHE
    Q6=+2 ..... SICHERHEITS-ABSTAND
    Q7=+20 ..... SICHERE HOEHE
    Q8=+0 ..... RUNDUNGSRADIUS
    Q9=-1 ..... DREHSINN
11 CYCL DEF 22.0 RAEUMEN
    Q10=-5 ..... ZUSTELL-TIEFE
    Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
    Q12=500 ..... VORSCHUB RAEUMEN
12 L X+0 Y+20 M8
13 L Z+2 R0 F MAX M99
14 CYCL DEF 23.0 SCHLICHTEN TIEFE
    Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
    Q12=500 ..... VORSCHUB RAEUMEN
15 CYCL CALL

16 CYCL DEF 24.0 SCHLICHTEN SEITE
    Q9=-1 ..... DREHSINN
    Q10=-5 ..... ZUSTELL-TIEFE
    Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
    Q12=500 ..... VORSCHUB RAEUMEN
    Q14=+0 ..... AUFMASS SEITE
17 CYCL CALL

18 L Z+100 R0 F MAX M2
    
```





UP

19 LBL 1
20 L X+0 Y+0 RL
21 FL AN+0
22 FCT X+60 DR- R6
23 FLT AN+270
24 FL X+48 Y-16 AN+198
25 L X+8
26 RND R6
27 L Y-75
28 RND R6
29 FL AN+0
30 FCT DR+ R6
31 FLT AN+125
32 FCT DR- R5 CCX+33
33 FLT X+60 Y-64 AN+305
34 L Y-96
35 RND R7,5
36 L X-60
37 RND R7,5
38 L Y-64
39 FL AN+55
40 FCT DR- R5 CCX-33
41 FLT AN+235
42 FCT Y-75 DR+ R6
43 FLT AN+0
44 FCT X-8 DR+ R6
45 FLT AN+90
46 FCT Y-16 DR+ R6
47 FLT X-48 Y-16 AN+180
48 FL X-60 AN+162
49 FL AN+90
50 FCT X-54 Y+0 DR- R6
51 FSELECT 2
52 L X+0
53 LBL 0

54 LBL 2
55 L X+100 Y+20 RR
56 L Y-120
57 L X-100
58 L Y+20
59 L X+100
60 LBL 0
61 END PGM 7174 MM








```

0 BEGIN PGM 75108 MM
1 BLK FORM 0.1 Z X-80 Y-10 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5
4 TOOL CALL 1 Z S1000
5 L X-100 Y+0 R0 F MAX M3
6 L Z-20 F MAX
7 L X-80 Y+0 RL F600
8 FPOL X+0 Y+0
9 FC DR- R80 CCX+0 CCY+0
10 FCT DR- R8 CCPR+72 CCPA+120
11 FCT DR+ R64 CCX+0 CCY+0
12 FCT DR+ R8 CCY+8
13 FSELECT 2
14 FCT DR- R48 CCX+0 CCY+0
15 FCT DR- R8 CCPR+40 CCPA+120
16 FCT DR+ R32 CCX+0 CCY+0
17 FCT DR+ R6 CCY+6
18 FSELECT 2
19 FCT X+0 Y+20 DR- R20 CCX+0 CCY+0
20 FLT X+12 Y+20
21 FL AN-60
22 FCT Y+0 DR+ R4
23 FLT AN+0
24 FCT DR+ R4
25 FLT AN+120 PAR21 DP16
26 FCT DR- R8 CCY+34
27 FLT AN-60 PAR25 DP16
28 FCT Y+0 DR+ R4
29 FLT Y+0 AN+0
30 FCT DR+ R4
31 FLT AN+120 PAR27 DP16
32 FCT DR- R8 CCY+64
33 FLT AN-60 PAR31 DP16
34 FCT X+100 Y+0 DR+ R4
35 FSELECT 2
36 FL X+100 Y-20
37 FL X-80 Y-20
38 FL X-80 Y+0
39 END PGM 75108 MM

```



HAUPTPROGRAMM

```

0 BEGIN PGM 7176 MM
1 ..... FINGER MIT 20ER ZYKLEN
2 BLK FORM 0.1 Z X-85 Y-25 Z-12
3 BLK FORM 0.2 X+105 Y+75 Z+0
4 TOOL DEF 5 L+0 R+5
5 TOOL CALL 5 Z S2500
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 KONTUR
8 CYCL DEF 14.1 KONTURLABEL 1 /2
9 CYCL DEF 20.0 KONTUR-DATEN
  Q1=-10 ..... FRAESTIEFE
  Q2=1 ..... BAHN-UEBERLAPPUNG
  Q3=+1 ..... AUFMASS SEITE
  Q4=+1 ..... AUFMASS TIEFE
  Q5=+0 ..... KOORDINATEN OBERFLAECHE
  Q6=-2 ..... SICHERHEITS-ABSTAND
  Q7=+20 ..... SICHERE HOEHE
  Q8=+0 ..... RUNDUNGSRADIUS
  Q9=-1 ..... DREHSINN
10 CYCL DEF 22.0 RAEUMEN
  Q10=-10 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
11 L X-100 Y-20 M99

12 CYCL DEF 23.0 SCHLICHTEN TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
13 L M99

14 CYCL DEF 24.0 SCHLICHTEN SEITE
  Q9=-1 ..... DREHSINN
  Q10=-5 ..... ZUSTELL-TIEFE
  Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
  Q12=200 ..... VORSCHUB RAEUMEN
  Q14=+0 ..... AUFMASS SEITE
15 L M99

Freifahren, Ende 16 L Z+100 R0 F MAX M2

```



UP

17 LBL 1
18 L X-80 Y-20 RL
19 FL Y+0 AN+90
20 FCT DR- R80 CCA+120 CCX+0 CCY+0
21 FCT DR- R8
22 FCT DR+ R64 CCX+0 CCY+0
23 FCT DR+ R8 CCY+8
24 FSELECT 2
25 FCT DR- R48 CCA+120 CCX+0 CCY+0
26 FCT DR- R8
27 FCT DR+ R32 CCX+0 CCY+0
28 FCT DR+ CCY+8
29 FCT X+0 Y+20 DR- R20 CCA+90 CCX+0 CCY+0
30 FSELECT 2
31 FLT X+12 AN+0
32 FL Y+0 AN-60
33 RND R4
34 FL AN+0
35 RND R4
36 FL AN+120 PAR31 DP16
37 FCT DR- R8 ICCA-180 CCY+34
38 FLT Y+0
39 RND R4
40 FL AN+0
41 RND R4
42 FL PAR39 DP16
43 FSELECT 2
44 FCT DR- R8 ICCA-180 CCY+64
45 FLT Y+0
46 RND R4
47 FL X+100 AN+0
48 FL Y-20 AN+270
49 FL X-80 AN+180
50 LBL 0

51 LBL 2
52 L X-90 Y-30 RR
53 L Y+90
54 L X+110
55 L Y-30
56 L X-90
57 LBL 0
58 END PGM 7176 MM




Schema: **FK Kamm als Insel original und in Y gespiegelt, mit SL I Ausräumen**

Vorbereitung	<i>BLK FORM</i>	
	<i>TOOL 1...</i>	
Auflistung	<i>CYCL DEF 14 LABEL 1/2</i>	Teilkonturen
Arbeitszyklus	<i>CYCL DEF 6 AUSRAEUMEN</i>	
	<i>L X... Y... R0 F9999 M3</i>	
Ausführungen	<i>L Z... M99</i>	Obere Hälfte
	<i>CYCL DEF 7 NULLPUNKT</i>	
	<i>X0 Y-...</i>	
	<i>CYCL DEF 8 SPIEGELN Y</i>	
	<i>CYCL CALL</i>	Untere Hälfte Y-Werte gespiegelt
Freifahren, Ende	<i>L Z20 F9999 M2</i>	

UP, Teilkonturen	<i>LBL 1</i>	Insel original
	<i>L X... Y... RL</i>	
	<i>⋮</i>	
	<i>LBL 0</i>	

	<i>LBL 2</i>	Tasche original
	<i>L X... Y... RR</i>	
	<i>⋮</i>	
	<i>LBL 0</i>	



**HAUPTPROGRAMM**

```

0 BEGIN PGM 75122 MM
1 BLK FORM 0.1 Z X-80 Y-95 Z-50
2 BLK FORM 0.2 X+100 Y+75 Z+0
3 TOOL DEF 1 L+0 R+3
4 TOOL CALL 1 Z S200
5 CYCL DEF 14.0 KONTUR
6 CYCL DEF 14.1 KONTURLABEL 1 /2
7 CYCL DEF 6.0 AUSRAEUMEN
8 CYCL DEF 6.1 ABST -2 TIEFE -30
9 CYCL DEF 6.2 ZUSTLG -10 F100 AUFM +0
10 CYCL DEF 6.3 WINKEL +0 F200
11 L X-100 Y+0 R0 F MAX M3
12 L Z+2 F9999 M99
13 CYCL DEF 7.0 NULLPUNKT
14 CYCL DEF 7.1 X+0
15 CYCL DEF 7.2 Y-20
16 CYCL DEF 8.0 SPIEGELN
17 CYCL DEF 8.1 Y
18 CYCL CALL

19 CYCL DEF 8.0 SPIEGELN
20 CYCL DEF 8.1

Freifahren, Ende
21 L Z+20 F MAX M2

```

UP


22 LBL 1
23 L X-80 Y+0 RL
24 FPOL X+0 Y+0
25 FC DR- R80 CCX+0 CCY+0
26 FCT DR- R8 CCPR+72 CCPA+120
27 FCT DR+ R64 CCX+0 CCY+0
28 FCT DR+ R8 CCY+8
29 FSELECT 2
30 FCT DR- R48 CCX+0 CCY+0
31 FCT DR- R8 CCPR+40 CCPA+120
32 FCT DR+ R32 CCX+0 CCY+0
33 FCT DR+ R6 CCY+6
34 FSELECT 2
35 FCT X+0 Y+20 DR- R20 CCX+0 CCY+0
36 FLT X+12 Y+20
37 FL AN-60
38 FCT Y+0 DR+ R4
39 FLT AN+0
40 FCT DR+ R4
41 FLT AN+120 PAR37 DP16
42 FCT DR- R8 CCY+34
43 FLT AN-60 PAR41 DP16
44 FCT Y+0 DR+ R4
45 FLT Y+0 AN+0
46 FCT DR+ R4
47 FLT AN+120 PAR43 DP16
48 FCT DR- R8 CCY+64
49 FLT AN-60 PAR47 DP16
50 FCT X+100 Y+0 DR+ R4
51 FSELECT 2
52 FL X+100 Y-20
53 FL X-80 Y-20
54 FL X-80 Y+0
55 LBL 0

56 LBL 2
57 L X-90 Y-20 RR
58 L Y+100
59 L X+120
60 L Y-20
61 L X-90
62 LBL 0
63 END PGM 75122 MM



Aufgabe: Mehrfache Verwendung einer Insel-Kontur

Programm(e): _____



Vorbereitung

```
BLK FORM / TOOL ...
CYCL DEF 14 KONTUR-
LABEL 10/2/3/4/5
CYCL DEF 20.0 KONTUR-DATEN
CYCL DEF 22.0 RAEUMEN
L X... Y... M3
L Z... M99
```

Ohne LBL 1 = Original

Aufruf

Bearbeitung

Freifahren, Ende

L Z100 M2

Teilkonturen, UP

LBL 10 ... RR ... LBL 0

Tasche als äußere Begrenzung

LBL 1

```
NULLPUNKT = 0
DREHUNG = 0/MASSFaktor = 1
SPIEGELN = NO ENT
```

LBL 0

Insel 1, Original, nicht direkt verwendet
Zyklen rücksetzen

LBL 2

```
NULLPUNKT = ...
```

CALL LBL 1

LBL 0

Insel 2

LBL 3

```
NULLPUNKT/MASSFaktor = ...
```

CALL LBL 1

LBL 0

Insel 3

LBL 4

```
NULLPUNKT/DREHUNG = ...
```

CALL LBL 1

LBL 0

Insel 4

LBL 5


```
NULLPUNKT/SPIEGELN ...
```

CALL LBL 1

LBL 0

Insel 5





HAUPTPROGRAMM

```

0 BEGIN PGM 7177 MM
1 ..... MEHRFACHE VERWENDUNG
2 ..... DERSELBEN KONTUR-INSEL IN
3 ..... DER TASCHE
4 BLK FORM 0.1 Z X+0 Y+0 Z-20
5 BLK FORM 0.2 X+100 Y+80 Z+0
6 TOOL DEF 1 L+0 R+5
7 TOOL CALL 1 Z S3000
8 CYCL DEF 14.0 KONTUR
9 CYCL DEF 14.1 KONTURLABEL 10 /2 /3 /4 /5
10 CYCL DEF 20.0 KONTUR-DATEN
    Q1=-10 ..... FRAESTIEFE
    Q2=1 ..... BAHN-UEBERLAPPUNG
    Q3=+0 ..... AUFMASS SEITE
    Q4=+0 ..... AUFMASS TIEFE
    Q5=+0 ..... KOORDINATEN OBERFLAECHE
    Q6=-2 ..... SICHERHEITS-ABSTAND
    Q7=+20 ..... SICHERE HOEHE
    Q8=+0 ..... RUNDUNGSRADIUS
    Q9=-1 ..... DREHSINN
11 CYCL DEF 22.0 RAEUMEN
    Q10=-5 ..... ZUSTELL-TIEFE
    Q11=100 ..... VORSCHUB TIEFENZUSTELLUNG
    Q12=500 ..... VORSCHUB RAEUMEN
12 L X+10 Y+10 R0 F MAX M3
13 L Z+20 F9999 M99

Freifahren, Ende
14 L Z+100 F MAX M2
    
```

UP10, TASCHE

15 LBL 10
 16 L X-5 Y-5 RR
 17 L X-5 Y+85
 18 L X+105 Y+85
 19 L X+105 Y-5
 20 L X-5 Y-5
 21 LBL 0

UP1, ORIGINAL

22 LBL 1
 23 L X+0 Y+0 RL
 24 L X+18 Y+20
 25 L X+18 Y+0
 26 L X+0 Y+0
 27 CYCL DEF 7.0 NULLPUNKT..... ALLES RUECKSETZEN
 28 CYCL DEF 7.1 X+0
 29 CYCL DEF 7.2 Y+0
 30 CYCL DEF 10.0 DREHUNG
 31 CYCL DEF 10.1 ROT+0
 32 CYCL DEF 11.0 MASSFAKTOR
 33 CYCL DEF 11.1 SCL1
 34 CYCL DEF 8.0 SPIEGELN
 35 CYCL DEF 8.1
 36 LBL 0


37 LBL 2
 38 CYCL DEF 7.0 NULLPUNKT
 39 CYCL DEF 7.1 X+10
 40 CYCL DEF 7.2 Y+10
 41 CALL LBL 1
 42 LBL 0

43 LBL 3
 44 CYCL DEF 7.0 NULLPUNKT
 45 CYCL DEF 7.1 X+45
 46 CYCL DEF 7.2 Y+5
 47 CYCL DEF 11.0 MASSFAKTOR
 48 CYCL DEF 11.1 SCL1,5
 49 CALL LBL 1
 50 LBL 0

51 LBL 4
 52 CYCL DEF 7.0 NULLPUNKT
 53 CYCL DEF 7.1 X+60
 54 CYCL DEF 7.2 Y+45
 55 CYCL DEF 10.0 DREHUNG
 56 CYCL DEF 10.1 ROT+15
 57 CALL LBL 1
 58 LBL 0

59 LBL 5
 60 CYCL DEF 7.0 NULLPUNKT
 61 CYCL DEF 7.1 X+35
 62 CYCL DEF 7.2 Y+55
 63 CYCL DEF 8.0 SPIEGELN
 64 CYCL DEF 8.1 X
 65 CALL LBL 1
 66 LBL 0
 67 END PGM 7177 MM





- Arbeitsweise:**
- 2½ D-Schnitt in Z/X
 - Zustellung in Y

- Werkzeuge:**
- Schaftfräser R10 zum Schruppen
 - Schaftfräser R10 zum Schlichten

Schema:

Kniehebelgelenk mit 1 Werkzeug, konstanter Gleichlauf

Vorbereitung

Schruppen

```
BLK FORM
TOOL DEF ... R10
TOOL 2 ...
L Z50 R0 F ... M3
CYCL DEF 7.0 NULLPUNKT
CYCL DEF 7.1 X+Q108 Y0 / Z0
```

```
CALL LBL 1
```

Rechts

```
CYCL DEF 7.0 NULLPUNKT
CYCL DEF 7.1 X-Q108 Y100 / Z0
CYCL DEF 10.0 ROT180
```

```
CALL LBL 1
```

Links

Freifahren, Ende

```
L Z ... M2
```

Profil Z/X, UP

```
LBL 1
```

```
L X+55 Y-4 R0 F ... M3
L Z ...
```

Start Ebene
Start Z

```
LBL 2
```

```
L IY ... F100
L X ... Z ... F250
L X ... Z ...
CT X ... Z ...
L Z ... F9999
L X ... Z ...
L Z ... F ...
```

Y-Takt

Tangente
Bogen
Freifahren
Für weitere Schritte

```
CALL LBL 2 REP ...
```

```
L Z+ ...
CYCL DEF 7.0 NULLPUNKT
CYCL DEF ... X0 Y0 Z0
CYCL DEF ROT 0
```


Freifahren fertig
Rücksetzen

```
LBL 0
```



Lösung:

Kniehebelgelenk mit 1 Werkzeug, konstanter Gleichlauf



HAUPTPROGRAMM

```
0 BEGIN PGM 761260 MM
1 BLK FORM 0.1 Z X-50 Y+0 Z-52
2 BLK FORM 0.2 X+50 Y+100 Z+0
3 L Z+50 R0 F9999 M6
4 TOOL DEF 2 L+0 R+10
5 TOOL CALL 2 Z S4000
6 L Z+50 R0 F MAX M3
7 CYCL DEF 7.0 NULLPUNKT
8 CYCL DEF 7.1 X+Q108
9 CYCL DEF 7.2 Y+0
10 CALL LBL 1 ..... RECHTS

11 CYCL DEF 7.0 NULLPUNKT
12 CYCL DEF 7.1 X-Q108
13 CYCL DEF 7.2 Y+100
14 CYCL DEF 10.0 DREHUNG
15 CYCL DEF 10.1 ROT+180
16 CALL LBL 1 ..... LINKS

17 L Z+50 R0 F MAX M2
```

Freifahren, Ende



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



761260/3

Lösung:

Kniehebelgelenk mit 1 Werkzeug, konstanter Gleichlauf

UP, PROFIL Z/X

18 LBL 1
19 L X+55 Y-4 R0 F9999 M3
20 L Z-51 F1000

21 LBL 2
22 L IY+2 F100
23 L X+39 Z-51 F250
24 L X+28,354 Z-20,2
25 CT X+0 Z+0
26 L Z+5 R0 F9999
27 L X+55 Z+5
28 L Z-51 F1000
29 CALL LBL 2 REP 50/50

30 L Z+5 R0 F MAX
31 CYCL DEF 7.0 NULLPUNKT
32 CYCL DEF 7.1 X+0
33 CYCL DEF 7.2 Y+0
34 CYCL DEF 7.3 Z+0
35 CYCL DEF 10.0 DREHUNG
36 CYCL DEF 10.1 ROT+0
37 LBL 0
38 END PGM 761260 MM



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



761260/4

Schema:

Kniehebelgelenk mit 2 Werkzeugen, konstanter Gleichlauf, Erweiterung Schruppen

Vorbereitung
Schruppen

```
BLK FORM  
TOOL DEF ... R10  
TOOL 1 ...  
FN 1: Q1 = Q108 + 5  
L Z50 R0 F ... M3  
CYCL DEF 7.0 NULLPUNKT  
CYCL DEF ... X+Q1 Y0 Z+5
```

X > WZ-Radius

```
CALL LBL 1
```

```
CYCL DEF 7.0 NULLPUNKT  
CYCL DEF ... X-Q1 Y100 Z+5  
CYCL DEF 10.0 ROT180
```

```
CALL LBL 1
```

Fertigschnitt schon
vorhanden

```
TOOL 2 ...  
•  
•
```

```
CALL LBL 1
```

```
•  
•
```

```
CALL LBL 1
```

Freifahren, Ende

```
L Z ... M2
```

Profil Z/X, UP1
schon vorhanden

```
LBL 1
```

```
•  
•  
•  
•  
•  
•  
•  
•
```

```
LBL 0
```



Lösung:

Kniehebelgelenk mit 2 Werkzeugen, konstanter Gleichlauf

HAUPTPROGRAMM

Werkzeug 1

```
0 BEGIN PGM 761261 MM
1 BLK FORM 0.1 Z X-50 Y+0 Z-52
2 BLK FORM 0.2 X+50 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S2800
5 FN 1: Q1 = +Q108 + +5
6 L Z+50 R0 F9999 M3
7 CYCL DEF 7.0 NULLPUNKT
8 CYCL DEF 7.1 X+Q1
9 CYCL DEF 7.2 Y+0
10 CYCL DEF 7.3 Z+5
11 CALL LBL 1 ..... RECHTS

12 CYCL DEF 7.0 NULLPUNKT
13 CYCL DEF 7.1 X-Q1
14 CYCL DEF 7.2 Y+100
15 CYCL DEF 7.3 Z+5
16 CYCL DEF 10.0 DREHUNG
17 CYCL DEF 10.1 ROT+180
18 CALL LBL 1 ..... LINKS
```

Werkzeug 2

```
19 TOOL DEF 2 L+0 R+10
20 TOOL CALL 2 Z S4000
21 L Z+50 R0 F MAX M6
22 CYCL DEF 7.0 NULLPUNKT
23 CYCL DEF 7.1 X+Q108
24 CYCL DEF 7.2 Y+0
25 CALL LBL 1 ..... RECHTS

26 CYCL DEF 7.0 NULLPUNKT
27 CYCL DEF 7.1 X-Q108
28 CYCL DEF 7.2 Y+100
29 CYCL DEF 10.0 DREHUNG
30 CYCL DEF 10.1 ROT+180
31 CALL LBL 1 ..... LINKS
```

Freifahren, Ende

```
32 L Z+50 R0 F MAX M2
```

UP, PROFIL Z/X

```
33 LBL 1
34 L X+55 Y-4 R0 F9999 M3
35 L Z-51 F1000

36 LBL 2
37 L IY+2 F100
38 L X+39 Z-51 F250
39 L X+28,354 Z-20,2
40 CT X+0 Z+0
41 L Z+5 R0 F9999
42 L X+55 Z+5
43 L X+55 Z-51 F MAX
44 CALL LBL 2 REP 50/50

45 L Z+5 R0 F MAX
46 CYCL DEF 7.0 NULLPUNKT
47 CYCL DEF 7.1 X+0
48 CYCL DEF 7.2 Y+0
49 CYCL DEF 7.3 Z+0
50 CYCL DEF 10.0 DREHUNG
51 CYCL DEF 10.1 ROT+0
52 LBL 0
53 END PGM 761261 MM
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx D02



761261/3

Aufgabe: **Kegelstumpf stehend, 2½ D,
außen**


Programm(e): _____



- Arbeitsweise:**
- 2½ D-Schnitt
 - Höhenlinien in X/Y
 - Zustellung in Z

- Werkzeuge:**
- Schafffräser

Kegelstumpf stehend, 2D, horizontal, mit Vollkreisen



Berechnung der Schrittweiten pro Umlauf:

$$\Delta PR = \frac{\text{Endradius} - \text{Anfangsradius}}{\text{Schrittzahl}}$$

$$\Delta Z = \frac{\text{Endhöhe} - \text{Anfangshöhe}}{\text{Schrittzahl}}$$

Beispiel: Schrittzahl = 10

$$\Delta PR = \frac{35 - 15}{10} = \frac{20}{10} = 2$$

$$\Delta Z = \frac{-20 - 0}{10} = \frac{-20}{10} = -2$$



Schema:

**Kegelstumpf stehend, außen,
2½ D-Schnitt, horizontal**

Vorbereitung

Schruppen

<i>BLK FORM TOOL 1... R10 L Z20 R0 F9999 M3 CC X... Y... LP PR65 PA0 R... F... M... L Z+2</i>
<i>LBL 1</i>
<i>LP IPR-... RL F... L Z-30 CP IPA-360 DR- RL F... M98</i>
<i>CALL LBL 1 REP...</i>
<i>L Z... M6</i>

Z-Sicherheit

Schlichten

<i>TOOL 2... R10 L Z+... R0 F... M3 LP PR15 PA0 R0 F... LP IPR+Q108</i>
<i>L Z0 F150</i>
<i>LBL 2</i>
<i>LP IPR+1 L IZ-1 F200 CP IPA-360 DR- F1000</i>
<i>CALL LBL 2 REP...</i>

WZ-Radius außen
kompensieren
Aufsetzen

} Konuswinkel 45°
für Schräge
Vollkreis in X/Y


Freifahren, Ende

L Z20... M2



Lösung:

Kegelstumpf stehend, $2\frac{1}{2} D$ horizontal, Außenform mit Schaftfräser



```
0 BEGIN PGM 7162 MM
1 ..... STEHENDER KEGELSTUMPF 2½ D
2 ..... HORIZ., SCHAFTFRAESER
3 BLK FORM 0.1 Z X+0 Y+10 Z-31
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 TOOL DEF 1 L+0 R+10
6 TOOL CALL 1 Z S2000
7 CC X+50 Y+50
8 L Z+20 R0 F9999 M3
9 LP PR+65 PA+0
10 L Z+2

11 LBL 1
12 LP IPR-5 RL F200
13 L Z-30
14 CP IPA-360 DR- RL F300 M98
15 CALL LBL 1 REP 5/5
16 L Z+20 R0 F9999 M6


17 TOOL DEF 2 L+0 R+10
18 TOOL CALL 2 Z S2800
19 L Z+20 R0 F9999 M3
20 LP PR+15 PA+0 R0 F9999 M3
21 LP IPR+Q108
22 L Z+2
23 L Z+0 F150

24 LBL 2
25 LP IPR+1
26 L IZ-1 F200
27 CP IPA-360 DR- F1000 M98
28 CALL LBL 2 REP 19/19

29 L Z+20 R0 F9999 M2
30 END PGM 7162 MM
```

Aufgabe: Kegelstumpf stehend, 3D, Außenform

Programm(e): _____




- Arbeitsweise:**
- 3D-Schnitt von oben nach unten
 - Rotation um Nullpunkt im Kegelstumpf-Zentrum

- Werkzeuge:**
- Schaftfräser R6

Detail:

Kegelstumpf stehend, 3D, vertikal, Außenform mit Schaftfräser, Radiuskompensation



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



7282/2

Schema:

Kegelstumpf stehend, 3D, Außenform

Vorbereitung

```
BLK FORM
TOOL ...
L Z ... M3
CYCL DEF 7.0 NULLPUNKT ...
Q3 = 15 + Q108
Q4 = 35 + Q108
```

Nullpunkt in Teilemitte
WZ-Kompensation

1 Aufruf

```
CALL LBL 1
```

Freifahren, Ende

```
L Z ... M2
```

1 Umlauf, UP

```
LBL 1
```

```
L XQ3 Y0 R0 F9999 M3
```

Anfahren auf komp.
X-Zeichnungs-Wert

```
L Z2
```

```
L Z0 F100
```

Aufsetzen

```
LBL 2
```

```
L XQ4 Y0 Z-20
```

3D-Schnitt,
F bleibt frei!
VS für 1. Schnitt = 100
Z zurück

```
L Z0 F9999
```

```
L XQ3 Y0 F2000
```

```
CYCL DEF 10.0 DREHUNG
```

```
CYCL DEF 10.1 IROT-5
```

```
L Y0 F500
```

Vorschub für
Folgeschnitte

```
CALL LBL 2 REP ...
```

```
CYCL DEF 10.0 DREHUNG
```


```
CYCL DEF 10.1 ROT+0
```

```
LBL 0
```



Lösung:

Kegelstumpf stehend, 3D vertikal, Außenform mit Schaftfräser, wiederholbar



HAUPTPROGRAMM

```
0 BEGIN PGM 7282 MM
1 ..... STEHENDER KEGELSTUMPF 3D
2 ..... VERTIKAL, SCHAFTFRAESER
3 ..... FERTIGSCHNITT
4 BLK FORM 0.1 Z X+10 Y+10 Z-32
5 BLK FORM 0.2 X+90 Y+90 Z+0
6 TOOL DEF 1 L+0 R+10
7 TOOL CALL 1 Z S4000
8 L Z+20 R0 F9999 M3
9 CYCL DEF 7.0 NULLPUNKT
10 CYCL DEF 7.1 X+50
11 CYCL DEF 7.2 Y+50
12 FN 1: Q3 = +15 + +Q108 ..... X OBEN
13 FN 1: Q4 = +35 + +Q108 ..... X UNTEN
14 CALL LBL 1 ..... 1. SCHNITT

15 L Z+50 R0 F9999 M2
```

UP

```
16 LBL 1
17 L X+Q3 Y+0 R0 F9999 M3 ..... ANFAHREN OBEN
18 L Z+2
19 L Z+0 F100

20 LBL 2
21 L X+Q4 Y+0 Z-20 ..... SCHNITT ABWAERTS
22 L Z+0 F9999 ..... Z FREI
23 L X+Q3 Y+0 F2000
24 CYCL DEF 10.0 DREHUNG
25 CYCL DEF 10.1 IROT-5 ..... FEINHEIT DER ROTATION
26 L Y+0 F500 ..... OBEN QUER
27 L F500
28 CALL LBL 2 REP 71/71 ..... ANZAHL DER ROTAT.-SCHRITTE

29 CYCL DEF 10.0 DREHUNG
30 CYCL DEF 10.1 ROT+0
31 LBL 0
32 END PGM 7282 MM
```




HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C05



7282/4




- Arbeitsweise:**
- 3 D-Schnitt von unten nach oben
 - Vorarbeiten mit 2 Kreistaschen

- Werkzeuge:**
- Schaftfräser R6

Lösung:

Kegelstumpf stehend, 3D vertikal, Innenform mit Schaftfräser, aufwärts



HAUPTPROGRAMM

```
0 BEGIN PGM 7283 MM
1 ..... STEHENDER KEGELSTUMPF 3D
2 ..... VERT., INNEN., SCHAFTFRAESER
3 ..... FERTIGSCHNITT B06
4 BLK FORM 0.1 Z X+0 Y+10 Z-32
5 BLK FORM 0.2 X+100 Y+90 Z+0
6 TOOL DEF 1 L+0 R+10
7 TOOL CALL 1 Z S2800
8 L Z+20 R0 F9999 M3
9 CYCL DEF 7.0 NULLPUNKT
10 CYCL DEF 7.1 X+50
11 CYCL DEF 7.2 Y+50
12 FN 1: Q3 = +35 + -Q108
13 FN 1: Q4 = +15 + -Q108
14 CALL LBL 1

15 L Z+50 R0 F9999 M2
```

UP

```
16 LBL 1 ..... ALLERERSTER SCHNITT
17 L X+10 Y+0
18 L Z-20 R0 F400 M3
19 L X+Q3 Y+0

20 LBL 2 ..... FOLGESCHNITTE 3D AUFWAERTS
21 L X+Q3 Y+0 Z-20 F1000
22 L X+Q4 F9999
23 L Z-30 F2000
24 CYCL DEF 10.0 DREHUNG
25 CYCL DEF 10.1 IROT-5
26 CALL LBL 2 REP 71/71

27 CYCL DEF 10.0 DREHUNG
28 CYCL DEF 10.1 ROT+0
29 LBL 0
30 END PGM 7283 MM
```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



7283/3

Lösung:

Kegelstumpf stehend, 3D vertikal, Innenform mit Schaftfräser, aufwärts, mit Schruppen

HAUPTPROGRAMM

```
0 BEGIN PGM 7284 MM
1 ..... STEHENDER KEGELSTUMPF 3D
2 ..... VERTIKAL, INNENFORM
3 ..... AUS DEM VOLLEN MIT
4 ..... SCHAFTFRAESER
5 BLK FORM 0.1 Z X+0 Y+10 Z-32
6 BLK FORM 0.2 X+100 Y+90 Z+0
7 TOOL DEF 1 L+0 R+10
8 TOOL CALL 1 Z S2800
9 L Z+20 R0 F9999 M3
10 FN 0: Q1 = +20 ..... TIEFE
11 FN 0: Q2 = +35 ..... RADIUS
12 CYCL DEF 5.0 KREISTASCHE
13 CYCL DEF 5.1 ABST -2
14 CYCL DEF 5.2 TIEFE -Q1
15 CYCL DEF 5.3 ZUSTLG -5 F150
16 CYCL DEF 5.4 RADIUS Q2
17 CYCL DEF 5.5 F500 DR+
18 L X+50 Y+50
19 L Z+2 M99 ..... SCHRUPPEN
20 FN 0: Q1 = +10
21 FN 0: Q2 = +15

22 L Z-18 M99
23 CYCL DEF 7.0 NULLPUNKT
24 CYCL DEF 7.1 X+50
25 CYCL DEF 7.2 Y+50
26 FN 1: Q3 = +15 + -Q108 ..... XA UNTEN
27 FN 1: Q4 = +35 + -Q108 ..... XB OBEN
28 FN 0: Q5 = -25 ..... Z TIEFE 1. SCHNITT

29 CALL LBL 1 ..... SCHRÄGE SCHRUPPEN

30 FN 0: Q5 = -30 ..... Z TIEFE 2.SCHNITT

31 CALL LBL 1 ..... SCHLICHTEN

32 L Z+50 R0 F9999 M2
```

UP, PROFIL

```
33 LBL 1
34 L X+Q3 Y+0 Z+Q5 R0 F400 M3 ..... A
35 L X+Q4 Y+0 Z+0 ..... B

36 LBL 2
37 L X+Q3 Y+0
38 L Z+Q5
39 L X+Q4 Y+0 Z+0 F1000 ..... B
40 L X+Q3 F9999 ..... A
41 L Z+Q5 F2000
42 CYCL DEF 10.0 DREHUNG
43 CYCL DEF 10.1 IROT-5
44 CALL LBL 2 REP 71/71
45 CYCL DEF 10.0 DREHUNG
46 CYCL DEF 10.1 ROT+0
47 LBL 0
48 END PGM 7284 MM
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C09



7284/3

Aufgabe: **Halbkugel, Innenbearbeitung,
2D, horizontal,**

Programm(e): _____




- Arbeitsweise:**
- Vollkreise (Höhenlinien) in X/Y
 - Vorarbeiten mit 2 Kreistaschen

- Werkzeuge:**
- Schruppen (Kreistaschen) mit Schafffräser R10
 - Kontur bearbeiten mit Kugelfräser R5

Schema:

Halbkugel, Innenbearbeitung mit Kugelfräser konventionell mit R0, 2D, horizontal



Schema: **Halbkugel, Innenbearbeitung mit Kugelfräser konventionell mit R0, 2D, horizontal, abwärts arbeitend**

Vorbereitung

*BLK FORM
TOOL 1 . . . R5
L Z10 R0 F9999 M3
CC X50 Y50
LP PR50 PA0 R0 F9999
LP IPR-Q108
L Z+0 R0 F100*

Z-Sicherheit
Ebene X/Y
 $PR = R_{\text{Teil}} - R_{\text{WZ}}$
WZ-Radius berücks.
Eintauchen

Anfahren

LBL 1

Vollkreise

*CC X50 Y50
CP DR+
CC X50 Z0
CP IPA2 DR+*

Ebene X/Y

Zustellung

Winkelschritte

Ebene Z/X


CALL LBL 1 REP 44/44

L Z+ . . . M2



Lösung:

Halbkugel, Innenbearbeitung mit Kugelfräser konventionell mit R0, 2D, horizontal




```

0 BEGIN PGM 7322 MM
1 BLK FORM 0.1 Z X-2 Y-2 Z-55
2 BLK FORM 0.2 X+102 Y+102 Z+0
3 TOOL DEF 1 L+0 R+5
4 TOOL CALL 1 Z S4000
5 L Z+70 R0 F9999 M6..... Z-SICHERHEIT
6 CC X+50 Y+50..... MITTELPUNKT IN X/Y
7 LP PR+50 PA+0 R0 F9999 M3..... STARTPUNKT
8 LP IPR-Q108..... WZ-RADIUS BERUECKSICHTIGEN
9 L Z+0 R0 F100

10 LBL 1
11 CC X+50 Y+50..... MITTELPUNKT IN X/Y
12 CP DR+ F500..... KREISRING
13 CC X+50 Z+0..... MITTELPUNKT IN Z/X
14 CP IPA+2 DR+ F100..... TIEFER FAHREN (KREISBOGEN)
15 CALL LBL 1 REP 44/44..... WEITERE UMLAEUFE

16 L Z+70 R0 F9999 M2
17 END PGM 7322 MM
    
```

Anzeigewert setzen



Lösung:

Halbkugel, Innenbearbeitung mit Schaft- und Gesenkfräser

HAUPTPROGRAMM

0 BEGIN PGM 7323 MM
1 BLK FORM 0.1 Z X-2 Y-2 Z-51
2 BLK FORM 0.2 X+102 Y+102 Z+0

Schaftfräser

3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 FN 0: Q1 = +45 OBERER TASCHENRADIUS
6 CALL LBL 1 SCHRUPPEN
7 L Z+100 R0 F MAX M6 WERKZEUGWECHSEL

Gesenkfräser

8 TOOL DEF 2 L+0 R+5
9 TOOL CALL 2 Z S4500
10 CALL LBL 2 SCHLICHTEN

Freifahren, Ende

11 L Z+100 R0 F MAX M2

UP1, Schruppen mit Schaftfräser

12 LBL 1
13 CC X+50 Y+50 MITTELPUNKT X/Y
14 LP PR+0 PA+0 R0 F MAX M3
15 CYCL DEF 5.0 KREISTASCHE
16 CYCL DEF 5.1 ABST -2
17 CYCL DEF 5.2 TIEFE -14
18 CYCL DEF 5.3 ZUSTLIG -5 F150
19 CYCL DEF 5.4 RADIUS Q1
20 CYCL DEF 5.5 F500 DR+
21 L Z+2 R0 F MAX M99
22 FN 0: Q1 = +30 UNTERER TASCHENRADIUS
23 L Z-12 R0 F MAX M99
24 LBL 0

UP2, Schlichten mit Gesenkfräser (Kugelform)

25 LBL 2
26 LP PR+50 PA+0 R0 F500 M3 STARTPUNKT
27 LP IPR-Q108
28 L Z+0 R0 F MAX


29 LBL 20
30 CC X+50 Y+50 MITTELPUNKT X/Y
31 C DR+ F500 KREISRING
32 CC X+50 Z+0 MITTELPUNKT Z/X
33 CP IPA+2 DR+ F150 TIEFER FAHREN
34 CALL LBL 20 REP 44/44 WEITERE KREISRINGE

35 LBL 0
36 END PGM 7323 MM



Aufgabe: **Halbkugel, Außenbearbeitung
mit Kugelfräser, abwärts
arbeitend, $2\frac{1}{2} D$, horizontal**

Programm(e): _____



- Arbeitsweise:**
- Vollkreise (Höhenlinien) in X/Y
 - Zustellung mit Winkelschritten in Z/X
 - Vorarbeiten z.B. mit Konturinsel oder Schraubenlinien-Interpolation

- Werkzeuge:**
- (Schruppen mit Schafffräser R5)
 - Schlichten mit Kugelfräser R5

Schema:

**Halbkugel, Außenbearbeitung mit Kugelfräser,
abwärts arbeitend 2½ D, horizontal**

Vorbereitung

Schichten

*BLK FORM
TOOL 1 . . . R5
L Z 20 R0 F9999 M3*

Aufruf

CALL LBL 1

Freifahren, Ende

L Z . . . M2

UP

<i>LBL 1</i>
<i>L X+50 Y+50 R . . . F . . . M . . . L Z+Q108 F100</i>
<i>LBL 2</i>
<i>CC Z-35 X+50 CP IPA+2 DR+ F100</i>
<i>CC X+50 Y+50 CP DR- F500</i>
<i>CALL LBL 2 REP . . .</i>
<i>LBL 0</i>

Mitte anfahren
Werkzeug-Radius
berücksichtigen

} Zustellen in Z/X


} Vollkreis in X/Y

Weitere Schnitte



Lösung:

Halbkugel, Außenbearbeitung mit Kugelfräser, abwärts arbeitend, $2\frac{1}{2} D$, horizontal



HAUPTPROGRAMM

```
0 BEGIN PGM 72841 MM
1 ..... AUSSENKUGEL 2D MIT
2 ..... KUGELFRAESER SCHNITTFOLGE
3 ..... ABWAERTS
4 BLK FORM 0.1 Z X+0 Y+10 Z-36
5 BLK FORM 0.2 X+100 Y+90 Z+5
6 TOOL DEF 1 L+0 R+5
7 TOOL CALL 1 Z S2800
8 L Z+20 R0 F9999 M3
9 CALL LBL 1
```

Freifahren, Ende

```
10 L Z+100 R0 F9999 M2
```

UP

```
11 LBL 1
12 L X+50 Y+50 ..... MITTELPUNKT X/Y
13 L Z+Q108 F100 ..... AUFSETZEN MIT WZ-RADIUS

14 LBL 2
15 CC Z-35 X+50
16 CP IPA+2 DR+ F100 ..... ZUSTELLEN
17 CC X+50 Y+50
18 CP DR- F500 ..... VOLLKREIS
19 CALL LBL 2 REP 44/44
20 LBL 0
21 END PGM 72841 MM
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C05



72841/3

Aufgabe: **Halbkugel, Außenbearbeitung
mit Schafffräser,
nur von unten, 3D**

Programm(e): _____



Arbeitsweise:


- 3D-Schnitt vertikal
- Stets von unten

Werkzeuge:

- Schafffräser R10

Schema:

Halbkugel, Außenbearbeitung mit Schafffräser, nur von unten, 3D, 1 Schnitt



Schema:

**Halbkugel, Außenbearbeitung mit Schafffräser,
nur von unten, 3D, 1 Schnitt**

Vorbereitung

Vorpositionieren

<i>BLK FORM TOOL 1 . . . L Z50 R0 F MAX M6</i>	
<i>L X120 Y50 R0 F MAX M3 CYCL DEF 7.0 NULLPUNKT X50 Y+50 Z-50 L Y0 Z0 R0 F MAX</i>	
<i>LBL 2</i>	
<i>CC Z . . . X . . . L Y0 . . . LP PR50 PA90 R0 . . .</i>	
<i>LBL 3</i>	
<i>LP IPA-5</i>	
<i>CALL LBL 3 REP 17/17</i>	
<i>L X50 . . . L X65 Z0 . . . CYCL DEF 10.0 DREHUNG CYCL DEF 10.1 IROT+5</i>	
<i>CALL LBL 2 REP 53/53</i>	
<i>CYCL DEF 7.0 NULLPUNKT = 0 CYCL DEF 10.0 DREHUNG = 0 L Z10 R0 F9999</i>	

Nullpunkt in
Kugelmitte

Kompens. Kreismitte
Kontur-Werte

„Unechter
Kreisbogen“


1 Winkelschritt
(Raumwinkel)

Freifahren
1 weiterer Bogen

Rücksetzen
Freifahren

Lösung:

Halbkugel, Außenbearbeitung mit Schafffräser, nur von unten, 3D, 1 Schnitt



```
0 BEGIN PGM 76130 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-55
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+120 Y+50 R0 F MAX M3
7 CYCL DEF 7.0 NULLPUNKT
8 CYCL DEF 7.1 X+50
9 CYCL DEF 7.2 Y+50
10 CYCL DEF 7.3 Z-50
11 L Y+0 Z+0 R0 F MAX ..... ENDTIEFE (KUGELZENTRUM)

12 LBL 2
13 CC Z+0 X+Q108 ..... SYSTEMATISCHER VERSATZ IN X
14 L Y+0 F200
15 LP PR+50 PA+90 R0 F200 ..... KUGELRADIUS

16 LBL 3
17 LP IPA-5 ..... 1 WINKELSCHRITT (RAUMWINK.)
18 CALL LBL 3 REP 17/17

19 L X+50 Y+0 R0 F2000
20 L X+65 Z+0 F1000
21 CYCL DEF 10.0 DREHUNG ..... 1 WEITERER BOGEN
22 CYCL DEF 10.1 IROT+5
23 CALL LBL 2 REP 53/53

24 CYCL DEF 7.0 NULLPUNKT
25 CYCL DEF 7.1 X+0
26 CYCL DEF 7.2 Y+0
27 CYCL DEF 7.3 Z+0
28 CYCL DEF 10.0 DREHUNG
29 CYCL DEF 10.1 ROT+0
30 L Z+20 R0 F MAX
31 END PGM 76130 MM
```



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C09



76130/4

Schema:

**Halbkugel, Außenbearbeitung mit Schafffräser,
nur von unten, 3D, 2 Schnitte**

Vorbereitung

*BLK FORM... / TOOL 1... R 10
L X... Y... R0 F9999 M3
CYCL DEF 7.0 NULLPUNKT
 X50 Y50 Z-50
L Y0 Z15...*

Außen vorpositionieren

Kugelzentrum

Schruppen

LBL 1

*CC Z15 XQ108
L Y0 F200
LP PR50 PA90...*

Z-Aufmaß

Schwenken Ebene
Kontur-Werte

LBL 2

1 Bogen, grob, nicht
bis auf höchsten
Winkel

LP IPA-10 F500

CALL LBL 2 REP 3/3

*L X50 Y0 F MAX
L X65 Z15 F1000
CYCL DEF 10 DREHUNG IROT+18*

Z-Aufmaß

Zusätzliche Bögen

CALL LBL 1 REP...

*CYCL DEF 10 DREHUNG ROT+0
L Y0 Z0 F...*

Schlichten

LBL 3

*CC Z0 XQ108
L Y0
LP PR50 PA90 F200*

Z-Sollmaß

1 Bogen, fein,
bis auf höchsten
Raumwinkel
zurückfahren

LBL 4

LP IPA-1 F800

CALL LBL 4 REP 89/89

*L X+50 Y+0...
L X+65 Z+0 F1000
CYCL DEF 10 DREHUNG IROT+1*

CALL LBL 3 REP...

*CYCL DEF 7 NULLPUNKT = 0
CYCL DEF 10 DREHUNG = 0*

Freifahren, Ende

L Z20... M2




HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C05



76131/2

**Halbkugel, Außenbearbeitung mit Schafffräser,
nur von unten, 3D, 2 Schnitte**



```

0 BEGIN PGM 76131 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-60
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+120 Y+50 R0 F MAX M3
7 CYCL DEF 7.0 NULLPUNKT
8 CYCL DEF 7.1 X+50
9 CYCL DEF 7.2 Y+50
10 CYCL DEF 7.3 Z-50
11 L Y+0 Z+15 R0 F MAX ..... Z-AUFMASS

12 LBL 1
13 CC Z+15 X+Q108 ..... Z-AUFMASS
14 L Y+0 F200
15 LP PR+50 PA+90 R0 F200 ..... KUGELRADIUS

16 LBL 2 ..... 1 BOGEN GROB
17 LP IPA-10 R0 F500
18 CALL LBL 2 REP 3/3

19 L X+50 Y+0 R0 F MAX
20 L X+65 Z+15 R0 F1000
21 CYCL DEF 10.0 DREHUNG ..... 19 ZUSAETZLICHE BOEGEN
22 CYCL DEF 10.1 IROT+18
23 CALL LBL 1 REP 19/19

24 CYCL DEF 10.0 DREHUNG
25 CYCL DEF 10.1 ROT+0
    
```

Schruppen



Lösung:

Halbkugel, Außenbearbeitung mit Schafffräser, nur von unten, 3D, 2 Schnitte


Schichten

```
26 L Y+0 Z+0 R0 ..... Z-SOLLMASS
27 LBL 3
28 CC Z+0 X+Q108 ..... Z-SOLLMASS
29 L Y+0
30 LP PR+50 PA+90 F200 ..... KUGELRADIUS
31 LBL 4 ..... 1 BOGEN FEIN
32 LP IPA-1 F800
33 CALL LBL 4 REP 89/89
34 L X+50 Y+0 F2000
35 L X+65 Z+0 F1000
36 CYCL DEF 10.0 DREHUNG ..... 71 WEITERE BOEGEN
37 CYCL DEF 10.1 IROT+1
38 CALL LBL 3 REP 359/359
39 CYCL DEF 7.0 NULLPUNKT
40 CYCL DEF 7.1 X+0
41 CYCL DEF 7.2 Y+0
42 CYCL DEF 7.3 Z+0
43 CYCL DEF 10.0 DREHUNG
44 CYCL DEF 10.1 ROT+0
45 L Z+20 R0 F MAX
46 END PGM 76131 MM
```



Aufgabe: **Gewinde fräsen,
dreigängig M64 x 1,5 innen,
2 Zustellungen**

Programm(e): _____



Vorbedingung: • Mitte muß frei sein!

Arbeitsweise: • Schraubenlinien-Interpolation in Ebene X/Y mit
simultaner Z-Bewegung
• Arbeitsrichtung abwärts

Werkzeuge: • Gewindefräser mit einer Schneide, die dem Gewindeprofil
entspricht

Schema: **Gewinde fräsen,
dreigängig M64 x 1,5 innen,
2 Zustellungen**

Vorbereitung

*BLK FORM
TOOL 1 . . .
L Z+20 R0 F9999 M3
L X+50 Y+50 . . .
CC*

Freifahren
Mitte anfahren
Auto. Pol-Übernahme

**1. Schnitt
mit Aufmaß**

LBL 1

*L Z+0 . . . F500
LP PR+31 PA0 RR F150*

Aufsetzen in Z
Radius-Aufmaß

LBL 2

CP IPA-360 IZ-4,5 DR- F500

Schraubenlinie

CALL LBL 2 REP 4/4

*LP PR0 PA0 R0 F9999
L Z+0*

**2. Schnitt
auf Fertigmaß**

LP PR+32 PA0 RR F150

Fertigmaß, nachsetzen

LBL 3

CP IPA-360 IZ-4,5 DR- F500

Schraubenlinie

CALL LBL 3 REP 4/4

*LP PR0 PA0 R0 F9999
CYCL DEF 10.0 DREHUNG
CYCL DEF 10.1 IROT+120*

Mitte

CALL LBL 1 REP 2/2


Weitere Gänge

Freifahren, Ende

L Z20 . . . M2

Lösung:

Gewinde fräsen, dreigängig M64 x 1,5 innen, 2 Zustellungen



```
0 BEGIN PGM 7164 MM
1 BLK FORM 0.1 Z X+0 Y+10 Z-20
2 BLK FORM 0.2 X+100 Y+90 Z+0
3 TOOL CALL 1 Z S2000
4 TOOL DEF 1 L+0 R+20
5 L Z+20 R0 F9999 M3
6 L X+50 Y+50 R0 F9999 M3
7 CC
```

1. Gang, beginnend bei 0°

```
8 LBL 1
9 L Z+0 R0 F9999
```

1. Schnitt, Durchmesser 62

```
10 LP PR+31 PA+0 RR F500 ..... WAND ANFAHREN
```

```
11 LBL 2
12 CP IPA-360 IZ-4,5 DR- F500 ..... SCHRAUBENLINIE
13 CALL LBL 2 REP 4/4
```

```
14 LP PR+0 PA+0 R0 F9999 ..... MITTE ANFAHREN
15 L Z+0
```

Fertigschnitt, Durchmesser 64

```
16 LP PR+32 PA+0 RR F500 ..... WAND ANFAHREN
```

```
17 LBL 3
18 CP IPA-360 IZ-4,5 DR- ..... SCHRAUBENLINIE
19 CALL LBL 3 REP 4/4
```

```
20 LP PR+0 PA+0 R0 F9999 ..... MITTE ANFAHREN
21 CYCL DEF 10.0 DREHUNG ..... ROTATION
```

```
22 CYCL DEF 10.1 IROT+120
```

2. und 3. Gang, je 120° versetzt

```
23 CALL LBL 1 REP 2/2
```

```
24 L Z+20 R0 M2
25 END PGM 7164 MM
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C05



7164/3

Aufgabe: Umriß fräsen mit mehreren Zustellungen

Programm(e): _____



Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____



**Konventionelle
Vorbereitung
Ladewerte**

BLK FORM

FN 0: Q1 = ...
FN 0: Q2 = ...
FN 0: Q3 = ...
FN 0: Q4 = ...
FN 0: Q5 = ...
FN 0: Q6 = ...

TOOL 1 ... R5
L Z+ ... R0 F9999 M3

Vorschub Z
 Vorschub Ebene
 Start-Z
 End-Z
 Anzahl Schnitte
 Abrundung

Ausführen

CALL LBL 1

Freifahren, Ende

L Z100 ... M2

UP

LBL 1

Startwerte

FN 2: Q12 = Q4 - Q3
FN 4: Q12 = Q12 DIV Q5
FN 0: Q13 = Q3

Zustellbereich Z
 Zustellschritt Z
 Z-Start = Z-Aktuell
 Unkorrigiert !

Startposition

L X-10 Y+70 R0 F9999 M3

LBL 2

Endposition

L ZQ13 FQ1
L X+10 Y+70 RL FQ2
RND ...
L ... / RND ... / L ...
L X ... Y ... RL
RND ...
L X- ... Y ... R0 ...

FN 1: Q13 = Q13 + Q12

Tief fahren
 Anfahren 1. Kont.-Pkt
 Weich anfahren

1./ Letzter Kontur-Pkt
 Weich wegfahren
 Wegfahren, unkorrig.

Z-Zustell. rechnen

FN 11: IF Q13 GT Q4
GOTO LBL 2


FN 9: IF Q13 EQU Q4
GOTO LBL 2

} Rücksprung /
 weiterer Schnitt,
 wenn Endtiefe noch
 nicht gefräst

L Z2 F9999

LBL 0





HAUPTPROGRAMM

```
0 BEGIN PGM 7285 MM
1 ..... UMFRAESEN MIT MEHREREN
   ZUSTELLUNGEN
2 BLK FORM 0.1 Z X-10 Y+0 Z-40
3 BLK FORM 0.2 X+100 Y+100 Z+0
4 FN 0: Q1 = +100 ..... VORSCHUB Z
5 FN 0: Q2 = +500 ..... VORSCHUB EBENE
6 FN 0: Q3 = -2 ..... START-Z
7 FN 0: Q4 = -20 ..... END-Z
8 FN 0: Q5 = +4 ..... ANZAHL SCHNITTE
9 FN 0: Q6 = +10 ..... ABRUNDUNG
10 TOOL DEF 1 L+0 R+5
11 TOOL CALL 1 Z S2800
12 L Z+10 R0 F9999 M3
13 CALL LBL 1
14 L Z+100 R0 F9998 M2
```

Freifahren, Ende




UP

```
15 LBL 1
16 FN 2: Q12 = +Q4 - +Q3
17 FN 4: Q12 = +Q12 DIV +Q5 ..... ZUSTELLUNG Z
18 FN 0: Q13 = +Q3 ..... Z AKTUELL
19 L X-10 Y+70 R0 F9998 M3

20 LBL 2
21 L Z+Q13 FQ1
22 L X+10 Y+70 RL FQ2
23 RND R2
24 L Y+90
25 RND RQ6
26 L X+50
27 RND RQ6
28 L X+90 Y+50
29 RND RQ6
30 L Y+10
31 RND RQ6
32 L X+50
33 RND RQ6
34 L X+10 Y+50
35 RND RQ6
36 L Y+70
37 RND R2
38 L X-10 Y+70 R0 F9998
39 FN 1: Q13 = +Q13 + +Q12
40 FN 11: IF +Q13 GT +Q4 GOTO LBL 2
41 FN 9: IF +Q13 EQU +Q4 GOTO LBL 2
42 L Z+2 F9998
43 LBL 0
44 END PGM 7285 MM
```





Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Schema:

Rechtecktasche fertig fräsen

Ladewerte

```
•  
•  
FN 0: Q1 = ... Q9 = ...  
BLK FORM  
TOOL 1 ... R5  
L Z10 R0 F9999 M3
```

```
CALL LBL 1
```

Freifahren, Ende

```
L Z ... M2
```

UP

```
LBL 1
```

```
FN 4: Q13 = Q3 DIV 2  
FN 4: Q14 = Q4 DIV 2  
FN 4: Q16 = Q108 DIV 4  
FN 4: Q17 = Q7 DIV 2
```

```
L X+Q1 Y+Q2 R0 F9999 M3
```

```
L Z+Q9
```

```
L Z-Q5 FQ7
```

```
L IXQ13 YQ2 RL
```

```
RND RQ ... FQ ...
```

```
L IY ...
```

```
•
```

```
•
```

```
L IX0 IYQ14
```

```
RND RQ ... FQ ...
```

```
L XQ1 YQ2 R0 F9999
```

```
L Z+Q9
```

```
LBL 0
```

X/2

Y/2

Anfahr-Radius (RND)

Anfahr-Vorschub (RND)

Mitte anfahren

Z-Sicherheit

Z-Tiefe


1. Konturpunkt

Letzter Konturpunkt

In Mitte fahren

Freifahren in Z





HAUPTPROGRAMM

```

0 BEGIN PGM 7286 MM
1 FN 0: Q1 = +50 ..... X-MITTE
2 FN 0: Q2 = +50 ..... Y-MITTE
3 FN 0: Q3 = +90 ..... LAENGE X
4 FN 0: Q4 = +70 ..... BREITE Y
5 FN 0: Q5 = +20 ..... TIEFE Z
6 FN 0: Q6 = +10 ..... ABRUNDUNG
7 FN 0: Q7 = +1000 ..... VORSCHUB Z
8 FN 0: Q8 = +300 ..... VORSCHUB EBENE
9 FN 0: Q9 = +2 ..... Z-SICHERHEIT
10 BLK FORM 0.1 Z X+0 Y+10 Z-40
11 BLK FORM 0.2 X+100 Y+90 Z+0
12 TOOL DEF 1 L+0 R+5
13 TOOL CALL 1 Z S2800
14 L Z+10 R0 F9998 M3
15 CALL LBL 1

Freifahren, Ende      16 L Z+50 R0 M2

```

Lösung:

Rechtecktasche fertig fräsen

UP

```
17 LBL 1
18 FN 4: Q13 = +Q3 DIV +2
19 FN 4: Q14 = +Q4 DIV +2
20 FN 4: Q16 = +Q6 DIV +4
21 FN 4: Q17 = +Q7 DIV +2
22 L X+Q1 Y+Q2 R0 F9998 M3
23 L Z+Q9
24 L Z-Q5 FQ7
25 L IX+Q13 Y+Q2 RL
26 RND RQ16 FQ17
27 L IY+Q14 IX+0 FQ8
28 RND RQ6 FQ17
29 L IX-Q3 IY+0
30 RND RQ6 FQ17
31 L IY-Q4 IX+0
32 RND RQ6 FQ17
33 L IX+Q3 IY+0
34 RND RQ6 FQ17
35 L IY+Q14 IX+0
36 RND RQ16 FQ17
37 L X+Q1 Y+Q2 R0 F9998
38 L Z+Q9
39 LBL 0
40 END PGM 7286 MM
```




HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C01



7286/4



Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____



HAUPTPROGRAMM

```

0 BEGIN PGM 7288 MM
1 ..... KREISZAPFEN UMFRAESEN
2 ..... MIT 2 SCHNITTEN UND
   ..... RADIUSZUSTELLUNG

3 FN 0: Q1 = +50 ..... X-MITTE
4 FN 0: Q2 = +50 ..... Y-MITTE
5 FN 0: Q3 = +0 ..... Z-OBERKANTE
6 FN 0: Q4 = +80 ..... DURCHMESSER
7 FN 0: Q5 = +1 ..... RADIUS-AUFMASS
8 FN 0: Q6 = +20 ..... FRAESTIEFE
9 FN 0: Q7 = +200 ..... FRAESVORSCHUB
10 BLK FORM 0.1 Z X+0 Y+10 Z-22
11 BLK FORM 0.2 X+100 Y+90 Z+0
12 TOOL DEF 1 L+0 R+10
13 TOOL CALL 1 Z S4000
14 L Z+10 R0 F9999 M3
15 CALL LBL 1

Freifahren, Ende
16 L Z+50 M2
    
```




UP

17	LBL 1	
18	FN 1: Q23 = +Q3 + +2	Z-VERFAHRHOEHE (SICHERHEIT)
19	FN 4: Q24 = +Q4 DIV +2	RADIUS
20	FN 1: Q25 = +Q24 + +Q5	RADIUS + AUFMASS
21	FN 1: Q34 = +Q25 + +Q108	RADIUS + AUFM. + WZ-RADIUS
22	FN 1: Q35 = +Q34 + +Q5	RAD. + 2 * AUFM. + WZ-RADIUS
23	FN 4: Q28 = +Q5 DIV +2	HALBES AUFMASS
24	FN 4: Q27 = +Q7 DIV +2	HALBER VORSCHUB
25	FN 3: Q29 = +Q7 * +2	DOPPELTER VORSCHUB
26	CC X+Q1 Y+Q2	
27	LP PR+Q35 PA+0 R0 F9999	UNKORRIG. STARTPUNKT
28	L Z+Q23	AUF Z-SICHERHEIT
29	L Z-Q6 FQ27	EINTAUCHEN
30	LP PR+Q25 PA+0 RL FQ7	ANFAHREN MIT AUFMASS
31	RND RQ28 FQ27	WEICH ANFAHREN MIT AUFM.
32	CP PA+0 DR- FQ7	KREIS MIT AUFMASS
33	RND RQ28 FQ27	WEGFAHREN MIT AUFMASS
34	LP PR+Q35 PA+0 R0 FQ29	
35	LP PR+Q24 PA+0 RL FQ29	FERTIGSCHNEIDEN
36	RND RQ28 FQ27	
37	CP DR- FQ7	
38	RND RQ28 FQ27	
39	LP PR+Q35 PA+0 R0 FQ29	FREIFAHREN
40	L Z+Q23 R0 F9999	
41	LBL 0	
42	END PGM 7288 MM	



Aufgabe: **1 Lochkreis parametrisch,
1 Werkzeug, 1 Zyklus
(Zentrieren)**

Programm(e): _____



Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

UP1

Startwerte
Vorbereitung

1. Bohrung

Schleifenbeginn

Ende UP1

<i>LBL 1</i>
<i>FN 0: Q24 = 1</i> <i>FN 0: Q25 = Q5</i> <i>FN 4: Q26 = 360 DIV Q4</i> <i>CC XQ1 YQ2</i> <i>LP PRQ3 PAQ5 R0 F9999 M3</i> <i>L Z+Q6 M99</i>
<i>LBL 2</i>
<i>FN 1: Q24 = Q24 +1</i> <i>FN 1: Q25 = Q25 +Q26</i> <i>LP PRQ3 PAQ25 M99</i> <i>FN 12: IF Q24 LT Q4 GOTO LBL 2</i>
<i>LBL 0</i>

Zähler
Aktueller Winkel
Winkelschritt

Pol
Anfahren

Zählen
Rechnen/Aktualisieren


Folgebohrungen

Kontrollieren mit
Rücksprung



Lösung:

1 Lochkreis parametrisch, 1 Werkzeug, 1 Zyklus (Zentrieren)



HAUPTPROGRAMM

```
0 BEGIN PGM 7416 MM
1 ..... 1 LOCHKREIS
2 BLK FORM 0.1 Z X+0 Y+10 Z-22
3 BLK FORM 0.2 X+100 Y+90 Z+0
Zentrieren
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 L Z+10 R0 F9999 M3
7 CYCL DEF 1.0 TIEFBOHREN ..... ZYKLUSTYP UND ZYKLUSWERTE
8 CYCL DEF 1.1 ABST -Q6
9 CYCL DEF 1.2 TIEFE -2
10 CYCL DEF 1.3 ZUSTLG -2
11 CYCL DEF 1.4 V.ZEIT 0
12 CYCL DEF 1.5 F100
Geometriedaten
13 FN 0: Q1 = +50 ..... X-MITTE
14 FN 0: Q2 = +50 ..... Y-MITTE
15 FN 0: Q3 = +20 ..... RADIUS
16 FN 0: Q4 = +10 ..... ANZAHL
17 FN 0: Q5 = +0 ..... STARTWINKEL
18 FN 0: Q6 = +2 ..... Z-SICHERHEIT, FERTIGUNGSW.
19 CALL LBL 1
Freifahren, Ende
20 L Z+20 R0 F9998 M2 ..... RUECKSPRUNG
```

UP, LOCHKREIS

```
21 LBL 1
22 FN 0: Q25 = +Q5
23 FN 0: Q24 = +1
24 CC X+Q1 Y+Q2
25 FN 4: Q26 = +360 DIV +Q4
26 LP PR+Q3 PA+Q5 R0
27 L Z+Q6 M99

28 LBL 2
29 FN 1: Q25 = +Q25 + +Q26
30 FN 1: Q24 = +Q24 + +1
31 LP PR+Q3 PA+Q25 M99
32 FN 12: IF +Q24 LT +Q4 GOTO LBL 2
33 LBL 0
34 END PGM 7416 MM
```



HEIDENHAIN


Fortgeschrittenen-Kurs TNC 4xx C05



7416/3

Aufgabe: 6 Lochkreise parametrisch

Programm(e): _____



Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

**Aufbau für 1 WZ /
1 Zyklus**

```

TOOL DEF 1 L0 R3,5
TOOL CALL ...

Q6 = +2      Abstand
Q7 = +2      Tiefe

TIEFBOHREN
ABST -Q6
TIEFE -Q7
    
```

**Vorbereitung für mehrere
WZ / Zyklen**

```

TOOL DEF 1 L0 R3,5
TOOL CALL ...

Q6 = +2      Abstand
Q7 = +2      Tiefe

TIEFBOHREN
ABST -Q6
TIEFE -Q7

CALL LBL 10
    
```

Weitere Bearb. / Zyklen
L Z100 F9999 M2

Geometrie

1. Lochkreis

```

Q1 = ... // Q5 = ...

CALL LBL 1
    
```

2. Lochkreis

```

Q3 = ... /

CALL LBL 1
    
```

3. Lochkreis

```

Q... = / Q... =

CALL LBL 1
    
```

6. Lochkreis

```

:
:
CALL LBL 1
    
```

```

LBL 10

Q1 = ... // Q5 = ...

CALL LBL 1

Q3 = ... /

CALL LBL 1

Q... = / Q... =

CALL LBL 1

:
:
CALL LBL 1

LBL 0
    
```

```

L Z100 M2

LBL 1
:
LP ... M3
L Z ... M99
LBL 3
:
:
LP PA M99

LBL 0
    
```


```

LBL 1
:
LP ... M3
L Z ... M99
LBL 3
:
:
LP PA M99

LBL 0
    
```

Lösung:

6 Lochkreise parametrisch, 1 Werkzeug, 1 Zyklus



HAUPTPROGRAMM

```
0 BEGIN PGM 74160 MM
1 ..... 6 LOCHKREISE
2 BLK FORM 0.1 Z X+0 Y+0 Z-22
3 BLK FORM 0.2 X+100 Y+100 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 L Z+10 R0 F9999 M3
7 FN 0: Q6 = +2
8 FN 0: Q7 = +2
9 CYCL DEF 1.0 TIEFBOHREN
10 CYCL DEF 1.1 ABST -Q6
11 CYCL DEF 1.2 TIEFE -Q7
12 CYCL DEF 1.3 ZUSTLG -2
13 CYCL DEF 1.4 V.ZEIT 0
14 CYCL DEF 1.5 F100
15 CALL LBL 10
16 L Z+50 R0 F MAX M2
```

Freifahren, Ende



HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C05



74160/3

Lösung:

6 Lochkreise parametrisch, 1 Werkzeug, 1 Zyklus

UP, LOCHKREIS GEOMETRIE

17 LBL 10
18 FN 0: Q1 = +50 X-MITTE
19 FN 0: Q2 = +50 Y-MITTE
20 FN 0: Q3 = +20 RADIUS
21 FN 0: Q4 = +10 ANZAHL
22 FN 0: Q5 = +0 STARTWINKEL
23 CALL LBL 1 1. LOCHKREIS

24 FN 0: Q3 = +30
25 CALL LBL 1 2. LOCHKREIS

26 FN 0: Q1 = +15
27 FN 0: Q2 = +15
28 FN 0: Q3 = +10
29 FN 0: Q4 = +8
30 CALL LBL 1 3. LOCHKREIS

31 FN 0: Q1 = +85
32 CALL LBL 1 4. LOCHKREIS

33 FN 0: Q2 = +85
34 CALL LBL 1 5. LOCHKREIS

35 FN 0: Q1 = +15
36 CALL LBL 1 6. LOCHKREIS
37 LBL 0

UP, DER LOCHKREIS SELBST

38 LBL 1 BEARBEITUNG
39 FN 0: Q25 = +Q5
40 FN 0: Q24 = +1
41 CC X+Q1 Y+Q2
42 FN 4: Q26 = +360 DIV +Q4
43 LP PR+Q3 PA+Q5 R0
44 L Z+Q6 M99

45 LBL 2
46 FN 1: Q25 = +Q25 + +Q26
47 FN 1: Q24 = +Q24 + +1
48 LP PR+Q3 PA+Q25 M99
49 FN 12: IF +Q24 LT +Q4 GOTO LBL 2

50 LBL 0
51 END PGM 74160 MM



Schema: **6 parametrische Lochkreise mit
3 WZ, 3 Zyklen, 2 Sicherheitshöhen**

Vorbereitung: *BLK-FORM*

Zentrieren

```

TOOL DEF 1 L0 R3,5
TOOL CALL ...
Q6 = +2
Q7 = +2
Q8 = +2
TIEFBOHREN
ABST-Q...
TIEFE-Q...
ZUSTLG-Q...

CALL LBL 10

```

Individuelle Werte:
ABSTAND
TIEFE
ZUSTELLUNG
Zyklus Def. 1

UP-Aufruf

Bohren

```

TOOL DEF 2 L0 R2,5
TOOL CALL ...
Q7 = +20
Q8 = +10

CALL LBL 10

```

Individuelle Werte:
TIEFE
ZUSTELLUNG

UP-Aufruf

Gewindebohren

```

TOOL DEF 3 L0 R3,0
TOOL CALL ...
Q... = +5
Q... = +15

GEWINDEBOHREN
ABST-Q...
TIEFE-Q...

CALL LBL 10

```

Individuelle Werte:
ABSTAND
TIEFE

Zyklus Def. 2

UP-Aufruf

Rücksprung *L Z100 M2*

Unterprogramme

```

LBL 10
Q1 = ... // Q5 = ...
CALL LBL 1
Q3 = ... // Q5 = ...
CALL LBL 1
:
LBL 0

```

```

LBL 1
:
LP ... M3
L Z... M99
LBL 3
:
LP PA M99
LBL 0

```


Lösung:

6 Lochkreise parametrisch, 3 Werkzeuge, 3 Zyklen

Zentrieren

```
0 BEGIN PGM 74161 MM
1 ..... 6 LOCHKR. M. 3 WZ U. 3 ZYKLEN
2 BLK FORM 0.1 Z X+0 Y+0 Z-22
3 BLK FORM 0.2 X+100 Y+100 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 FN 0: Q6 = +2 ..... ABSTAND
7 FN 0: Q7 = +2 ..... TIEFE
8 FN 0: Q8 = +2 ..... ZUSTELLUNG
9 L Z+10 R0 F9999 M3
10 CYCL DEF 1.0 TIEFBOHREN
11 CYCL DEF 1.1 ABST -Q6
12 CYCL DEF 1.2 TIEFE -Q7
13 CYCL DEF 1.3 ZUSTLG -Q8
14 CYCL DEF 1.4 V.ZEIT 0
15 CYCL DEF 1.5 F100
16 CALL LBL 10
17 STOP M6
```

Bohren

```
18 TOOL DEF 2 L+0 R+2,5
19 TOOL CALL 2 Z S3150
20 FN 0: Q7 = +20
21 FN 0: Q8 = +10
22 L Z+10 R0 F MAX M3
23 CALL LBL 10
24 STOP M6
```

Gewinden

```
25 TOOL DEF 3 L+0 R+3
26 TOOL CALL 3 Z S400
27 FN 0: Q6 = +5 ..... ABSTAND
28 CYCL DEF 2.0 GEWINDEBOHREN
29 CYCL DEF 2.1 ABST -Q6
30 CYCL DEF 2.2 TIEFE -15
31 CYCL DEF 2.3 V.ZEIT 0
32 CYCL DEF 2.4 F400
33 L Z+10 R0 F MAX M3
34 CALL LBL 10
35 L M30
```



Lösung:

6 Lochkreise parametrisch, 3 Werkzeuge, 3 Zyklen

Individuelle Maße der
Lochkreise

36 LBL 10
37 FN 0: Q1 = +50 X-MITTE
38 FN 0: Q2 = +50 Y-MITTE
39 FN 0: Q3 = +20 RADIUS
40 FN 0: Q4 = +10 ANZAHL
41 FN 0: Q5 = +0 STARTWINKEL
42 CALL LBL 1

43 FN 0: Q3 = +30
44 CALL LBL 1

45 FN 0: Q1 = +15
46 FN 0: Q2 = +15
47 FN 0: Q3 = +10
48 FN 0: Q4 = +8
49 CALL LBL 1

50 FN 0: Q1 = +85
51 CALL LBL 1

52 FN 0: Q2 = +85
53 CALL LBL 1
54 FN 0: Q1 = +15
55 CALL LBL 1

56 L Z+50 R0 F MAX M5
57 LBL 0

Lochkreise

58 LBL 1
59 FN 0: Q25 = +Q5
60 FN 0: Q24 = +1
61 CC X+Q1 Y+Q2
62 FN 4: Q26 = +360 DIV +Q4
63 LP PR+Q3 PA+Q5 R0
64 L Z+Q6 M99

65 LBL 2
66 FN 1: Q25 = +Q25 + +Q26
67 FN 1: Q24 = +Q24 + +1
68 LP PR+Q3 PA+Q25 M99
69 FN 12: IF +Q24 LT +Q4 GOTO LBL 2

70 LBL 0
71 END PGM 74161 MM





HEIDENHAIN

Fortgeschrittenen-Kurs TNC 4xx C01



74161/4



Q	Bedeutung	Wert	Anmerkung
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Vorbereitung

BLK FORM

Ladewerte

*FN 0: Q1 = 10
 FN 0: Q2 = 10
 FN 0: Q3 = 9
 FN 0: Q4 = 12
 FN 0: Q5 = 10
 FN 0: Q6 = 7
 FN 0: Q7 = 2*

X-Start
 Y-Start
 X-Abstand
 Y-Abstand
 Anzahl Spalten ||
 Anzahl Zeilen =
 Z-Sicherheit

TOOL 1.../CYCL1...ABST-Q7

CALL LBL 1

Freifahren, Ende

L Z50... M2

UP

LBL 1

Startwerte

*FN 0: Q21 = Q1
 FN 0: Q22 = Q2
 FN 0: Q25 = 0
 FN 0: Q26 = 0
 FN 0: Q23 = Q3*

X Aktuell
 Y Aktuell
 Aktuelle Spalte ||
 Aktuelle Zeile =
 X-Schritt kopieren

Schleifenbeginn

LBL 2

Positionieren

*L XQ21 YQ22 R0 F... M...
 L Z+Q7 M99
 FN 1: Q21 = Q21 + Q23
 FN 1: Q25 = Q25 + 1
 FN 12: IF Q25 LT Q5 GOTO LBL 2*

Bohren
 X-Schritt
 Spalte zählen


Neue Zeile

*FN 0: Q25 = 0
 FN 1: Q26 = Q26 + 1
 FN 0: Q23 = -Q23
 FN 1: Q21 = Q21 + Q23
 FN 1: Q22 = Q22 + Q4
 FN 12: IF Q26 LT Q6 GOTO LBL 2*

Spaltenzähler
 rücksetzen
 Zeile zählen
 X-Schritt umdrehen
 Kompensation
 Y-Schritt

LBL 0





HAUPTPROGRAMM

```
0 BEGIN PGM 746910 MM
1 ..... REGELMAESSIGE LOCHFLAECHE
2 ..... MIT Q-PARAMETERN
```

Ladedaten

```
3 BLK FORM 0.1 Z X+0 Y+0 Z-20
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 FN 0: Q1 = +10 ..... X-START
6 FN 0: Q2 = +10 ..... Y-START
7 FN 0: Q3 = +9 ..... X-ABSTAND
8 FN 0: Q4 = +12 ..... Y-ABSTAND
9 FN 0: Q5 = +10 ..... ANZAHL SPALTEN
10 FN 0: Q6 = +7 ..... ANZAHL ZEILEN
11 FN 0: Q7 = +2 ..... Z-SICHERHEIT
```

Freifahren, Ende

```
22 L Z+50 R0 F MAX M2
```

UP

24 LBL 1 VORBEREITUNGEN
 25 FN 0: Q21 = +Q1 X
 26 FN 0: Q22 = +Q2 Y
 27 FN 0: Q25 = +0 SPALTE
 28 FN 0: Q26 = +0 ZEILE
 29 FN 0: Q23 = +Q3 X-SCHRITT

Schleifenbeginn

30 LBL 2
 31 L X+Q21 Y+Q22 R0 F9999 POSITIONIEREN
 32 L Z+Q7 R0 F9999 M99 BOHREN
 33 FN 1: Q21 = +Q21 + +Q23 X-SCHRITT
 34 FN 1: Q25 = +Q25 + +1 SPALTE
 35 FN 12: IF +Q25 LT +Q5 GOTO LBL 2

Rücksprung

36 FN 0: Q25 = +0 SPALTENZAehler
 37 FN 1: Q26 = +Q26 + +1 ZEILE
 38 FN 0: Q23 = -Q23 X-RICHTUNG UMDREHEN
 39 FN 1: Q21 = +Q21 + +Q23 KOMPENSATION
 40 FN 1: Q22 = +Q22 + +Q4 Y-SCHRITT
 41 FN 12: IF +Q26 LT +Q6 GOTO LBL 2 UNFERTIG?
 42 LBL 0
 43 END PGM 746910 MM

